

القرية الهندسية
ENGINEERING VILLAGE
ELECTRONICS & MECHANICA
للإلكترونيات والميكانيكا

احترف

سكراتش 2.0

في 10 أيام

تعلم برمجة الألعاب والتطبيقات
بشكل عملي ، تطبيقي ، مبسط


تابعونا عبر قنوات التواصل الإجتماعي :

@EngineeringVillage

@EngineeringOman

Engineering Village

EngineeringVillageEV



SCRATCH






هذا الكتاب حر ومخصص للتوزيع المجاني في نسخته الإلكترونية ويمكنك أن تنشره بشكل كامل أو جزء منه بشرط ذكر المصدر.

جميع حقوق الطبع محفوظة لدى مؤسسة القرية الهندسية ولا يجوز استخدام الكتاب بأي صورة تجارية إلا بإذن خطي من المؤسسة وكل من يخالف ذلك فإنه يتحمل المساءلة القانونية.

الطبعة الثانية 2015





فريق العمل

الإشراف العام
فهد السيابي

فريق الإعداد
إسحاق الراشدي
سليم الراشدي
طارق المعمرى

التدقيق والمراجعة
وائل المغيرى

التصميم والإخراج الفني
حمزة السيابي
من فريق توباز
@TopazTeam





كلمة شكر

تتقدم إدارة مؤسسة القرية الهندسية بالشكر الجزيل والموصول لكل فرد كان له صلة وبصمة مميزة ساهمت في إثراء هذا العمل وإنجازه على أتم وجه

إدارة القرية الهندسية





المقدمة:

في ظل تطور البرمجة بشكل سريع بالعالم، أصبح تعليم البرمجة شيئاً أساسياً في المراحل المبكرة بسبب الصعوبات التي تواجه الطلاب في المرحلة الثانوية والمرحلة الجامعية في فهم لغات وتقنيات البرمجة وخاصة برمجة الكائنات. لذلك فإن التعليم المبكر والمبسط لهذا النوع من البرمجة يهيء الجيل الجديد للمرحلة القادمة .

وعلي هذا الأساس ظهرت لغات برمجية رسومية تساعد علي تعليم الأطفال والياfeعين البرمجة بمفهوم ممتع وسهل، ومن ضمن هذه اللغات لغة سكراتش.

سكراتش هي لغة برمجية رسومية تجعلك وبكل سهولة تنشأ قصصاً تفاعلية ، رسوم متحركة ، وألعاب مسلية ، صممت لمساعدة الياfeعين ومبتدئي البرمجة لتطوير مهارات التعلم. وتمكن المستخدم من إدخال صور وأصوات وفيديو وتغييرها.

في صفحات كتاب "سكراتش" سنتعلم المفاهيم البرمجية الأساسية كالتكرار والشروط والكائنات واللبنات. كما يحتوي الكتاب على العديد من الأمثلة الشيقة التي تساعد على فهم سكراتش بشكل سهل وممتع .

لقد كان لفريق **القرية الهندسية** تجارب سابقة في تدريس البرمجة للناشئة، وكم كانت سعادتنا كبيرة عند عودة بعض الطلبة ليلغونا بأنهم عند إنتقالهم للدراسة الجامعية إختاروا الدراسة في مجالات هندسية متعلقة بالبرمجة. هذا التأثير شجعنا على تأليف مناهج مبسطة تساعد الناشئة على تشكيل رؤاهم للمستقبل.

نترككم الآن مع كتاب سكراتش التطبيقي ونتمنى أن يكون له الأثر الأكبر في مستقبل الأجيال القادمة من المبرمجين العرب !

تحياتنا الحارة !
فريق العمل





الفهرس

- 9 - تمهيد . -----
- 14 - استخدام البرنامج . -----
- 16 - واجهة البرنامج . -----
- 18 - أساسيات البرمجة . -----
- 20 - استخدام الكائنات . -----
- 24 - لبنات البرمجة . -----
- 26 - استخدام الأوامر . -----
- 30 - حركة السيارة . -----
- 32 - طريقة حفظ المشروع . -----
- 34 - أمر التكرار المحدد . -----
- 38 - أمر التكرار المستمر . -----
- 40 - الوطواط الطائر . -----
- 44 - الأشكال الهندسية . -----
- 52 - برمجة أكثر من كائن . -----
- 54 - الحياة البحرية . -----
- 60 - تحدي الأرقام . -----
- 68 - حياة المدينة . -----
- 74 - إضافة تحدي لـ "حياة المدينة" . -----
- 76 - تحدي الملاعب / لعبة التنس . -----
- 82 - تحدي الأصوات . -----
- 86 - تحدي المزرعة . -----
- 90 - تحدي الفضاء . -----
- 98 - تحدي الحصن . -----
- 106 - تحدي الصحراء . -----
- 114 - العودة إلى الأرض . -----
- 126 - ملحق الرسم . -----
- 130 - ملحق تعاريف القوانين . -----





من الجميل أن نراكم مبرمجين منذ الصغر
#المبرمجون_الصغار





هل أنت جاهز للإنتلاق إلى عالم سكراتش الرائع ؟





تمهيد

انظر إلى العالم من حولك ! الهاتف الذي بيدك. السيارة التي تركبها. جهاز التكييف الذي يبرد الغرفة التي أنت فيها. الغسالة التي تغسل ملابسك الجميلة والرجال الآليين الذين تراهم في التلفاز. كلها عبارة عن آلات صنعها الإنسان للقيام بوظيفة محددة. أي أنها تتبع مجموعة معينة من التعليمات والأوامر.

خذ على سبيل المثال الهاتف النقال:

هناك من قام ببرمجته على الأمر التالي: عندما يقوم شخص ما بالاتصال، قم أيها الهاتف بتشغيل نغمة الرنين.

هناك من قام ببرمجة السيارة على الأمر التالي: عندما يوجد جسم ثابت خلف السيارة عند عودة السيارة للوراء، قومي أيتها السيارة بإصدار صوت الإنذار.

هناك من قام ببرمجة جهاز التكييف على الأمر التالي: عندما أضغط على الزر الأحمر من الريموت، إعمل أيها المكيف. وعندما أضغط عليه مرة أخرى، إنطفيء أيها المكيف.

هناك من قام ببرمجة جهاز الحاسوب على تلقي الحروف من لوحة المفاتيح وتلقي الأوامر من زر الفأرة.

إذن، نحن البشر نحتاج إلى لغة للتخاطب مع الأجهزة الإلكترونية (الغير حية!)، تسمى هذه اللغة: البرمجة

ويسمى الشخص الذي يقوم بكتابة التعليمات والأوامر البرمجية: المبرمج
في هذا الكتاب سوف تبدأ مشوارك معنا لتصبح مبرمجاً .. هيا بنا نبدأ





ما هي البرمجة ؟

ويكيبيديا: هي عملية كتابة تعليمات وتوجيه أوامر لجهاز الحاسوب أو أي جهاز آخر، لتوجيه هذا الجهاز وإعلامه بكيفية التعامل مع البيانات أو كيفية تنفيذ سلسلة من الأعمال المطلوبة.

لماذا نتعلم البرمجة ؟

عالمنا اليوم يحتاج إلى المزيد من المبرمجين المبدعين. في كل يوم نسمع عن هاتف جديد وتطبيقات رائعة في الهواتف الذكية وبرامج الحاسب الآلي ومواقع التواصل الإجتماعي. عندما تصبح مبرمجا تتمكن من المشاركة في تطوير العالم سواء عن طريق كتابة برامج الحاسوب أو تطبيقات الهواتف الذكية أو برمجة الأجهزة والمتحكمات في المشاريع والإبتكارات الإلكترونية.



ما هي لغات البرمجة ؟


وكما أن البشر يتواصلون بعدة لغات مختلفة كالعربية والصينية والفرنسية. فإن الأجهزة الإلكترونية تتواصل فيما بينها ومع الإنسان بلغات مختلفة تسمى لغات البرمجة. في هذا الكتاب سنتعرف على إحدى أبسط هذه اللغات كبداية وهي البرمجة بلغة سكراتش.





ما هو سكراتش ؟

عندما لاحظ الخبراء في الشركات الكبرى مثل جوجل ومايكروسوفت أنهم بحاجة إلى المزيد من المبرمجين المهرة، قرروا أن يبسطوا تعليم البرمجة ويوصلوها إلى طلبة المدارس حتى يتم تهيئة المبرمجين منذ صغر سنهم للمشاركة في تطوير العالم. وبالتعاون بين هذه الشركات الكبرى وبعض الجامعات الأكاديمية مثل جامعة MIT في الولايات المتحدة الأمريكية وغيرها، تم التوصل إلى إنتاج برنامج سكراتش الذي يبسط مبادئ البرمجة للناشئة. وفي هذا الكتاب محاولة منا لتقديمها للبناء في الوطن العربي ليواكبوا العالم في هذا المجال.






هدف الكتاب وطريقة استخدامه

هذا الكتاب التطبيقي يهدف إلى تعريف الناشئة بعالم البرمجة الممتع من خلال شرح المفاهيم عبر الأمثلة، وهو يتبع أسلوب التعليم من خلال الممارسة والتسلية. في هذه المغامرة سنبدأ بالتعرف على واجهة البرنامج ثم ننتقل خطوة بخطوة للتعرف على أوامر البرمجة المختلفة وتطبيقاتها الرائعة. يمكن تثبيت البرنامج في الحاسوب أو استخدامه مباشرة (بدون تثبيت) من موقع سكراتش الإلكتروني. شمر عن ساعدك ولننطلق.









إستخدام البرنامج

هناك طريقتان لاستخدام برنامج سكراتش :

الطريقة الأولى : عن طريق موقع سكراتش (Scratch.mit.edu) بدون تثبيت في جهاز الحاسوب، كما هو واضح في الصور

scratch.mit.edu

Scratch

تسجيل الدخول انضم إلى سكراتش

أُنشئ قصصاً وألعاباً ورسوماً متحركة شاركها مع الآخرين حول العالم

بعد فتح الموقع اضغط على زر "أنشئ"

انضم إلى سكراتش (لها تعليمات)

تعالف أسئلة

تعالف سكراتش

مجتمع تعلم إبداعى تمت فيه مشاركة 7383648 مشروعاً.

عن سكراتش | للترتيب للمتابعة

المشاريع المميزة

Bumperz بواسطة Rant

Ethereal بواسطة Nova

One player Tic-Tac-Toe بواسطة Prinseskat


Draw a snowflake بواسطة shoresbeep

geometry puzzle بواسطة Bobo927






سوف يظهر لك البرنامج ويمكنك استخدامه بحرية



الطريقة الثانية: لإستخدامه في جهازك يمكنك تنزيل البرنامج من الوصلة التالية:
Scratch.mit.edu/scratch2download
واتبع الخطوات الموضحة في الصور لتثبيت البرنامج.



إضغط على كلمة continue كما هو واضح في الصورة

في حال واجهتك أي مشكلة في التثبيت يرجى زيارة صفحة البرمجة للأطفال على موقع القرية الهندسية
www.ev-center.com

2




1

بعد تنزيل البرنامج اضغط على ايقونة التثبيت

انتظر حتى ينتهي التحميل بعدها افتح البرنامج بشكل تلقائي


3






واجهة البرنامج

بعد إنتهاءك من تثبيت البرنامج (أو عند فتحه مباشرة من الموقع الإلكتروني) سوف تظهر أمامك واجهة البرنامج. تعرف أكثر على مكونات البرنامج من خلال الدليل التالي:



- 1- نافذة الأوامر: حيث يتم تركيب وتجميع الأوامر.
- 2- شريط ل 3 نوافذ (الأوامر ، المظاهر و الأصوات).
- 3- نافذة اللبنة الأساسية (Blocks): لتصنيف الأوامر الرئيسية.
- 4- نافذة لبنات الاوامر: هنا نجد جميع الأوامر البرمجية.
- 5- شريط الأدوات: حيث يمكننا عمل نافذة جديدة، تغيير اللغة ، حفظ العمل ، طلب مساعدة وايضا قص ولزق.
- 6- شريط التحكم بنافذة العرض: تشغيل البرنامج و إيقافه و تكبير الشاشة وتصغيرها.
- 7- نافذة العرض: لمشاهدة العمل النهائي.
- 8- نافذة الكائنات : إختيار وإضافة وحذف الكائنات.
- 9- نافذة إعدادات الخلفية: للتحكم في الخلفية وتغييرها.





للنجاح

ضع قدمك
على أول الطريق
وعينك عند آخره





أساسيات البرمجة

في البرمجة بشكل عام نحتاج إلى ثلاث اشياء :
1- الكائن 2- الخصائص 3- المواصفات

مثال : السيارة




المواصفات:

الخصائص:

الكائن:


فيراري
330 كم / ساعة
أحمر
6.5
100,000 ريال



الاسم
السرعة
اللون
قوة المحرك
القيمة

سيارة سباق





مثال : جهاز الحاسوب


المواصفات:

الخصائص:

الكائن:

توشيبا ساتالايت

لابتوب
14 أنش
i7
فضي



الاسم
النوع
الحجم
المعالج
اللون

جهاز حاسوب



مثال : هاتف محمول


قم بتكملة المثال التالي حسب ما هو موضح.

المواصفات:

الخصائص:

الكائن:

.....
.....
.....
.....
.....



.....
.....
.....
.....
.....

.....






إستخدام الكائنات

الكائنات : هي الوحدات والأشكال التي سنقوم بتحريكها والعمل عليها واللعب بها.

أمثلة على الكائنات : حيوانات ، سيارات ، شخصيات كرتونية، والكثير من الصور الأخرى.
الكائنات هي أول ما تحتاج إلى إضافته قبل كتابة أي برنامج وهي التي تقوم بتنفيذ الأوامر والتعليمات التي نكتبها

هنا سنتعلم كيف نستخدم الكائنات في برنامج سكراتش (تنزيل، وإنشاء وتصوير)




لإختيار كائن من البرنامج

لرسم كائن جديد

لإختيار كائن من ملفات الحاسوب

لإختيار كائن باستخدام الكاميرا





عند إختيارك لكائن جديد من البرنامج يظهر لك التالي:



ثم اضغط OK

في حال اختيار رسم كائن جديد يكون له اسم تلقائي في نافذة الكائنات




هناك ملحق خاص بالرسم في آخر الكتاب .




عند اختيارك كائن باستخدام الكاميرا ستظهر كما في الصورة أمامك ، إضغط Save لحفظ الصورة.




في حال اختيار كائن من ملف بالحاسوب :





عند اختيارك لكائنات جديدة يظهر لك كل ما اخترته في نافذة الكائنات:



لحذف كائن ،
اضغط عليه بالزر
الأيمن ثم اضغط
delete



أيضا يمكنك تغيير الخلفية بنفس أسلوب اختيار الكائنات ولكن نتحكم بها من نافذة اعدادات الخلفية.

يمكنك اختيار الخلفية من ملف أو رسمها أو من البرنامج أو حتى باستخدام الكاميرا. جرب !

جرب إضافة الكائنات والخلفيات كما بالصورة .






لبينات البرمجة

في الصفحة السابقة تعرفنا على الكائنات. ولكن نحن لا نضع الكائنات في البرنامج لكي ننظر إليها ! نحن نضعها لكي نحركها ونتحكم فيها من خلال لينات البرمجة والتي تسمى "الأوامر" لأنها تأمر الكائنات بتنفيذ ما نريده منها ... تعرف أكثر على لينات البرمجة .

في لعبة تركيب المكعبات نركب الاشكال كالتالي:



1- نبدأ بوضع أول مكعب من اللعبة




2- نضعهن جنب بعض



3- حتى نقوم ببناء المبنى كاملاً


كذلك برمجة سكراتش ، تبني على مكعبات (لبينات) الأوامر







مثال: الهاتف النقال

عندما يأتي (اتصال) يظهر (الاسم) و نسمع (نغمة الاتصال) ويبدأ الجاهز في (الأهتزاز):



مثال لترتيب اللبنت على البرنامج





إستخدام الأوامر

عند تقريب أي أمر من آخر في نافذة الأوامر فإن الأوامر تلتصق مع بعضها أو تتداخل أيضاً ، لاحظ الأمثلة التالية:

مثال التلاصق عند تقريب أمر من آخر :



مثال التداخل عند تقريب أمر من آخر :




مثال عن الإحتواء :







مثال يجمع عمليات التلاصق والتداخل والإحتواء لبعض الأوامر



أنا متأكد من أنك ستستمتع في برمجة سكراتش أيها المبرمج .










لنبدأ في تعلم البرمجة ن
قم بفتح صفحة جديدة من برنامج سكراتش.

إضغط على File



إختر كائن جديد، وليكن (سيارة) . قم بحذف أي كائن آخر



من نافذة اللبنة الأساسية لنضغط على لبنة الحركة motion كما هو موضح في الصورة، المسؤول عن حركة الكائنات.





الآن سنتعلم سحب وإسقاط اللبنة كما في الصورة:

move 10 steps

إضغط على اللبنة وانتبه لحركة السيارة

كرر الضغط ،، قم بتغيير الخطوات بالرقم بداخل الدائرة من 10 إلى أي رقم آخر. لاحظ الفرق

move 20 steps

move 5 steps


يمكنك مضاعفة الأوامر بالضغط على الزر الأيمن على اللبنة:






طريقة حفظ المشروع

عند الإنتهاء من أي مشروع من مشروعاتك فإن لديك طريقتين لحفظ المشروع:
الطريقة الأولى: حفظ المشروع في الجهاز، بحيث تستخدمه كل مرة عن طريق جهازك الخاص وذلك كما هو موضح في الصورة التالية.




إختر الإسم المناسب ثم إضغط على كلمة (حفظ باسم) أو Save as في حال رغبتك في إعادة فتح مشروعك من جديد قم بالخطوات الموضحة في الصورة.






الطريقة الثانية: حفظ المشروع على موقع سكراتش ويمكنك مشاركة زملائك وجميع الناس بالمشروع سواء عن طريق الموقع أو المشاركة في مواقع التواصل الاجتماعي. تابع الخطوات التالية.



يجب أن يكون لديك حساب بموقع سكراتش لترفع المشروع على الموقع. إذا لم تمتلك حسابا، يمكنك بكل سهولة التسجيل من موقع سكراتش على الإنترنت وبعدها الرجوع إلى مشروعك ورفع مرة أخرى.

بعد رفع المشروع في الموقع، يمكنك الحصول عليه ضمن ملفاتك الخاصة، كما يمكنك نسخ وصلة المشروع لمشاركتها مع من تحب كما هو موضح في الصورة



أختار أحد خيارات المشاركة






عند طلبنا لكرة واحدة بأن تتحرك بمقدار 10 خطوات فإننا ندرج الأمر كالتالي :

move 10 steps



ولكن عند طلبنا لكرة أن تتحرك عشرة خطوات (مكررة) أربعة مرات فإننا نعمل كما هو موضح في الأوامر :

move 10 steps
move 10 steps
move 10 steps
move 10 steps



أو بشكل مبسط يمكنك كتابة الأمر الذي يأمر بالتكرار 4 مرات repeat

1 اضغط على Control

2 اختر الخيار repeat

3 اسحب إلى نافذة الأوامر


4 قم بإدخال الأمر الأول move في أمر التكرار



مثال: نعود لمثال السيارة لنضيف الحركة التكرارية. قم بتغيير رقم عدد التكرار وشاهد ماذا يحدث.



عند وصول السيارة لنهاية الإطار ، كيف نجعلها تواصل الحركة؟ نضيف الأمر التالي على الأمر السابق والذي نجده من ضمن أوامر Motion. **if on edge, bounce**. والذي يأمر الكائن بالارتداد عند نهاية الإطار.



جرب الآن واضغط كل مرة على الاوامر أو زد قيمة عدد التكرار. ماذا تلاحظ؟




ستنقلب السيارة عند الرجوع

ستنقلب السيارة عند الرجوع من نافذة الكائنات. اتبع الخطوات لحل المشكلة، ثم جرب :





لاحظ بأن جميع ألوان الأوامر الرئيسية هي بنفس ألوان الأوامر المستخدمة






النجاح ليس عدم
فعل الأخطاء




النجاح هو عدم
تكرار الأخطاء





إذا اردنا من إشارة المرور العمل مدى الحياة، فلا يمكننا استخدام أمر التكرار لعدد محدد من المرات، هنا يمكننا أن نأمر إشارة المرور بالعمل مدى الحياة باستخدام الأداة التالية:



لنعود لمثال السيارة ولنجعلها تتحرك طوال الوقت بدون توقف كما فعلنا مع إشارة المرور.



بعد استبدال repeat ب forever ماذا تلاحظ؟

السيارة لا تتوقف عن الحركة ابداً

معلومة :

كلمة forever تعني " إلى الأبد "





لجعل السيارة تتحرك بعد صدور حدث الضغط على العلم عندما نضغط على العلم تعمل الأوامر.


نعمل ذلك بالخطوات التالية:

1

إختر Events من قائمة الأوامر الأساسية




2 قم بنقل الامر التالي إلى نافذة الأوامر



3 قم بوضع الامر في أعلى الأوامر السابقة

3

الآن قم بالضغط على زر العلم المتواجد أعلى الشاشة:






في هذا الدرس سنستخدم الأوامر لنساعد الوطواط على الطيران.



اختر كائن الوطواط من نافذة الكائنات.
اكتب أوامر الحركة لجعله يتحرك ويرتد.
ماذا تلاحظ؟



إذا انعكس اتجاهه قم بضبط اتجاه الوطواط كما تعلمنا في الدرس السابق.



1 اختر looks من قائمة الأوامر الأساسية

2 قم بنقل الأمر التالي إلى نافذة الأوامر

next costume

3 لنضيف الأمر إلى الأوامر السابقة



4 اضغط على إشارة البداية





ماذا تلاحظ الآن؟ هل حركة الوطواط طبيعية؟

سنلاحظ بأنه يتحرك بسرعة، نريد أن نخفف سرعة ضرب الأجنحة وذلك عن طريق الأمر التالي:

1 إختار control من قائمة الأوامر الأساسية

2 قم بنقل الأمر التالي إلى نافذة الأوامر

3 قم بوضع الأمر في أسفل الأوامر السابقة

wait 1 secs

when green flag clicked

forever

move 10 steps

if on edge, bounce

next costume

wait 1 secs

wait 0.2 secs

سنلاحظ بأنه يتحرك ببطئ شديد، قم بتغيير الرقم الداخلي للأمر إلى 0.2

كيف ترى حركة الوطواط الآن؟ تحكم بسرعيته من خلال تغيير الرقم السابق. كيف تعمل حركة:

1 إختار costume من الشريط التالي

2 ستلاحظ وجود صورتين للوطواط

3 عند استخدام الأمر next costume يعمل على نقل تحريك الخفاش بالانتقال بين الصورتين

next costume

bat1-a 115x125

bat1-b 115x125





الوطواط يطير في خط مستقيم كما نلاحظ في المثال. نريد أن نجعله يرفرف في كل الزوايا. لنستخدم الحركة التالية:



1 قم بتغيير وضع الزاوية من هذا الخيار ثم شغل البرنامج ولاحظ الاختلاف

الآن لنضيف خط رسم للوطواط، لنستخدم خاصية القلم


1 إختار Pen من قائمة الأوامر الأساسية



2 إختار هذا الأمر لبدء الرسم

pen down

3 نضعه من ضمن الأوامر السابقة



لاحظ على شاشة العرض ماذا يحدث. سيبدأ الوطواط برسم خط خلفه. في حال إيقاف العرض وتشغيله مرة أخرى يبقى الرسم الأولي، حتى نستطيع مسح الرسم كل مرة والبدء من جديد نضيف الأمر التالي

clear





يمكنك تغيير لون الخط وحجمه بالأوامر التالية:

set pen color to

إضغط هنا لتغيير لون الخط وقم بإختيار اللون المناسب عن طريق تحريك مؤشر الفأرة على اللون المناسب في الشاشة

change pen size by 1

لتغيير حجم الخط

```
when clicked
clear
set pen color to
change pen size by 1
forever
  move 10 steps
  if on edge, bounce
  next costume
  wait 0.2 secs
  pen down
```

أمر تنظيف الشاشة عند البداية

أمر لون خط الرسم



أمر حجم خط الرسم

قم بإضافة هذا الأمر بالداخل وشاهد ما يحدث

change pen color by 10

ما هو الشكل الهندسي الذي يظهر معك ، قم بتجربة باقي الأوامر واستخدم كائنات مختلفة واستمتع بهذه الخاصية.

مثال آخر





الأشكال الهندسية

في هذا المكان سنتعلم كيف نرسم الأشكال الهندسية والتي في النهاية تخرجنا إلى أشكال نهائية جميلة. بعد نهاية هذا الدرس سنعمل على رسم أشكال كما في الرسم التالي.



سنبدأ بتعلم رسم الخط، اختر أي كائن تحب قم بتصغير الكائن حتى لا يأخذ مساحة من العرض

when clicked

go to x: 0 y: 0

pen down

repeat 10

move 10 steps

أمر البداية

أمر لكي نبدأ من منتصف الصفحة

أمر وضع القلم على الصفحة (تجده في الأمر الرئيسي Pen) بحيث يبدأ الرسم

نضيف هذا الأمر حتى نعمل على رسم خط مستقيم

قم بتجربة الأمر، ماذا لاحظت، الآن نريد أن نعيد الكرة ولكن بأن يعمل على مسح جميع ما سبق قبل البدء من جديد. سنقوم بإضافة الأمرين التاليين في أول الأوامر كالتالي

when clicked

clear

pen up

go to x: 0 y: 0

pen down

repeat 10

move 10 steps

أمر التنظيف

أمر رفع القلم





كما تعلم بأن المربع هو عبارة عن 4 خطوط مكررة ولكنها تقف على بعضها
إذا هيا لنرسم المربع

1

نضيف أمر الدوران
90 درجة لليمين

```
repeat 10
  move 10 steps
  turn 90 degrees
```

2

نكرر هذا الأمر أربع مرات ليعطينا
شكل مربع كما هو واضح




3

يمكن إختصار كل ذلك بأمر التكرار
المحدد ليعيد الأوامر المتشابهة 4 مرات



```
when clicked
  clear
  pen up
  go to x: 0 y: 0
  pen down
  repeat 4
    repeat 10
      move 10 steps
      turn 90 degrees
```




```
when clicked
  clear
  pen up
  go to x: 0 y: 0
  pen down
  repeat 10
    move 10 steps
  turn 90 degrees
  repeat 10
    move 10 steps
  turn 90 degrees
  repeat 10
    move 10 steps
  turn 90 degrees
  repeat 10
    move 10 steps
  turn 90 degrees
```



دعنا نستمتع مع المربعات، لنعمل على رسم 6 مربعات بشكل جميل باستخدام الأوامر التالية

```
when clicked
clear
pen up
go to x: 0 y: 0
pen down
repeat 6
  repeat 4
    repeat 10
      move 10 steps
    turn 90 degrees
  turn 60 degrees
```


إضافة أمر التكرار 6 مرات



نضيف أمر الدوران
60 درجة لليمين

```
when clicked
clear
pen up
go to x: 0 y: 0
change pen size by 2
set pen color to green
pen down
repeat 9
  repeat 4
    repeat 10
      move 10 steps
    turn 90 degrees
  turn 40 degrees
  change pen color by 10
```


قم بتجربة تغيير الزوايا وعدد التكرار ، مثلًا مكرر 9 مرات بزاوية 40. جرب وشاهد ماذا يحدث، أيضا قم بتغيير الألوان وحجم الخط كالتالي.





سنبدأ الآن عمل المعين الذين تكون أضلاعه على زاويتي 60 و 120 درجة.. أتبع المثال التالي وأختار كائن حسب رغبتك، قم بتصغير الكائن حتى يناسب شاشة العرض


```
when clicked
clear
pen up
go to x: 0 y: 0
pen down
repeat 2
  move 100 steps
  turn 60 degrees
  move 100 steps
  turn 120 degrees
```



تحرك بمقدرا 100 خطوة ثم التفاف 60 درجة، بعدها تحرك 100 خطوة ثم ألتفاف 120 درجة. نكرر بأمر التكرار مرتين.

```
when clicked
clear
pen up
go to x: 0 y: 0
pen down
repeat 3
  repeat 2
    wait 1 secs
    move 100 steps
    turn 60 degrees
    wait 1 secs
    move 100 steps
    turn 120 degrees
  turn 120 degrees
```


سنقوم بعمل تكرار لجميع ما سبق 3 مرات لنعمل شكلا جميلاً





كيف سيكون شكل النموذج الهندسي عند استخدام عدد مرات التكرار 6 مرات والتحرك بدوران 60 درجة. اختر لون مختلف وحجم للخط.

```
when clicked
clear
pen up
go to x: 0 y: 0
pen down
repeat 6
  repeat 2
    wait 0.1 secs
    move 100 steps
    turn 60 degrees
    wait 0.1 secs
    move 100 steps
    turn 120 degrees
  turn 60 degrees
pen up
move 200 steps
```




أمر رفع القلم عن شاشة العرض حتى يتوقف عن الكتابة
أمر التحرك بمقدار 200 خطوة عن الرسم


تحدي نفسك واكتشف أي الأشكال التالية هي للأوامر التالية



1



2



3





A

```

when clicked
  clear
  pen up
  go to x: 0 y: 0
  pen down
  set pen color to red
  set pen size to 2
  repeat 3
    repeat 2
      wait 0.1 secs
      move 50 steps
      turn 60 degrees
      wait 0.1 secs
      move 50 steps
      turn 120 degrees
    turn 120 degrees
  set pen color to blue
  turn 60 degrees
  repeat 3
    repeat 2
      wait 0.1 secs
      move 70 steps
      turn 60 degrees
      wait 0.1 secs
      move 70 steps
      turn 120 degrees
    turn 120 degrees
  pen up
  move 200 steps

```

B

```

when clicked
  clear
  pen up
  go to x: 0 y: 0
  pen down
  set pen color to red
  set pen size to 2
  repeat 3
    repeat 2
      wait 0.1 secs
      move 50 steps
      turn 60 degrees
      wait 0.1 secs
      move 50 steps
      turn 120 degrees
    turn 120 degrees
  set pen color to blue
  turn 60 degrees
  repeat 6
    repeat 2
      wait 0.1 secs
      move 30 steps
      turn 60 degrees
      wait 0.1 secs
      move 30 steps
      turn 120 degrees
    turn 60 degrees
  pen up
  move 200 steps

```

C

```

when clicked
  clear
  pen up
  go to x: 0 y: 0
  pen down
  set pen color to red
  set pen size to 2
  repeat 6
    repeat 2
      wait 0.1 secs
      move 50 steps
      turn 60 degrees
      wait 0.1 secs
      move 50 steps
      turn 120 degrees
    turn 60 degrees
  set pen color to blue
  turn 60 degrees
  repeat 6
    repeat 2
      wait 0.1 secs
      move 30 steps
      turn 60 degrees
      wait 0.1 secs
      move 30 steps
      turn 120 degrees
    turn 60 degrees
  pen up
  move 200 steps


```






كيف تعتقد بأنه يمكن رسم الكرة من خلال الأوامر. سنترك هذا التحدي الآن بيد الكائن الوطواط ليقوم بالمهمة. الدائرة هي عبارة عن دوران بمقدار 360 درجة

```
when clicked
clear
pen up
go to x: 0 y: 0
pen down
repeat 360
  move 1 steps
  turn 1 degrees
```



يمكننا عمل دائرة بداخل دائرة أو اشكال متلفة من الدوائر بألوان مختلفة وأحجام مختلفة


```
when clicked
clear
pen up
go to x: 0 y: 0
pen down
repeat 360
  set pen color to purple
  move 1 steps
  turn 1 degrees
repeat 360
  set pen color to blue
  move 1.5 steps
  turn 1 degrees
```





الآن لاحظ ما يمكننا عمله بمجموعة من الدوائر تتداخل فيما بينها

```
when clicked
clear
pen up
go to x: 0 y: 0
pen down
set pen size to 2
repeat 8
  repeat 360
    set pen color to purple
    move .5 steps
    turn 1 degrees
  repeat 360
    set pen color to orange
    move 1 steps
    turn 1 degrees
  turn 45 degrees
pen up
glide 1 secs to x: 197 y: -3
```






في الدروس السابقة كنا نقوم ببرمجة كائن واحد . في هذا الدرس سنقوم بتعلم كيفية برمجة أكثر من كائن .

يجب أن تعرف أيها المبرمج المعلومات التالية :

أولاً : كل كائن له صفحة خاصة للبرمجة .

ثانياً : يمكنك نسخ برمجة أي كائن إلى كائن آخر (سنقوم بتطبيقه لاحقاً).

ثالثاً : عند تشغيل البرنامج من العلم الأخضر فإنه ينفذ برمجة جميع الكائنات .




لاحظ أن المؤشر على كائن الخنفساء






في حالة اختارنا القطة ستظهر صفحة البرمجة فارغة لأن البرمجة السابقة كانت في صفحة الخفساء .



لاحظ أن المؤشر على كائن القطة

يمكنك البرمجة الآن في صفحة كائن القطة





الحياة البحرية

سنتعلم هنا كيف نعمل الحياة البحرية وكيف نجعل الكائنات البحرية تتحرك وتلون وتشكل.



1 قم بتنزيل كائنات بحرية من الكائنات المتوفرة في البرنامج

2 اختر خلفية معبرة لأعماق البحر




2 الآن، اضغط على الكائن حتى يصغر

1 اضغط على زر التصغير

الآن سنقوم بتوزيع الكائنات وسنجعل كل كائن يتحرك بصورة معينة





سنبدأ بالعمل مع هذه السمكة

```
when clicked
  forever
```

نختار عند الضغط أمر التكرار المستمر

```
when clicked
  forever
    move 3 steps
    if on edge, bounce
```

من أمر الحركة نختار سرعة الحركة 3 وأمر الإرتداد (عندما يصل إلى نهاية الإطار يرتد)



في حال ارتداد السمكة وانعكاسها نعيد تضبيط الإتجاه من خلال الزر التالي كما تعلمنا سابقا



الآن سنعمل على نسخ الأوامر السابقة لهذه السمكة

```
when clicked
  forever
    move 3 steps
    if on edge, bounce
    right
```

1 قم بسحب الأوامر من هذه النافذة

2 الآن، ضع الأوامر على الكائن الذي تريد نسخ الأوامر إليه ، ثم اضغط عليه لتتأكد أن الأوامر قد نسخت في نافذة أوامره.

```
change color effect by 25
```

من قائمة الأوامر الرئيسية نختار looks ثم نضيف هذا الأمر مع الأوامر السابقة للسمكة حتى نجعلها مضيئة


```
when clicked
  forever
    move 3 steps
    if on edge, bounce
    change color effect by 5
```

يمكنك أيضا تغيير سرعة السمكة حتى تختلف عن الأخرى





سنقوم ببرمجة نجم البحر الآن، سيجعله يكبر ويصغر وستتركه في مكان ما على القاع



سننتقل الآن لبرمجة سرطان البحر





2 إضغط على هذا الخيار
ليضعف الكائن ثم اختر
الكائن المرغوب

يمكنك مشاهدة ملحق الرسم للتعرف
أكثر على كيفية تعديل الصور




1 سنقوم بإضافة
مظهر جديد للكائن
حتى نجعله يمشي

5 ثم إضغط على هذا
الزر حتى نستطيع
التعديل على الرسم

4 إضغط على
المظهر الجديد

1 إختار هذه
الخاصية للقص



2 قم بتحديد الأرجل
وتحريكها ، قم بتقريب
الرجلين بالوسط لبعظهما
وإبعاد الإثنتين الأخرى

when clicked
forever
نختار عند الضغط
أمر التكرار المستمر

when clicked
forever
move 4 steps
if on edge, bounce
مع أمر ارتداد عند الحد
حركة الكائن 4 خطوات

when clicked
forever
move 4 steps
if on edge, bounce
next costume
wait 0.1 secs
نظهر المظهر التالي
مع وقت انتظار 0.1 ثانية





أخيرا ، نبرمج الأخطبوط
يمكنك نسخ برمجة سرطان البحر أو كتابة برمجة مشابهة كالتالي:



يمكنك أيضا تغيير بعض الأرقام
المتواجدة في الأوامر، كسرعة الحركة
والوقت بين اشكال مظاهر الكائن.



2 قم بتغيير زاوية حركة الأخطبوط بأي اتجاه عشوائي



3 تذكر إستخدام هذه الخاصية

استمتع الآن بمشاهدة الحياة البحرية، كما يمكنك إضافة انواع أخرى من الكائنات وبرمجتها.





هذه أهد نهايات العمل بإضافة بعض الكائنات



تحدي الأرقام

في هذا التحدي سنساعد بكتابة الأرقام التي تعرض مع الصوت، فعندما يظهر لنا الرقم العشوائي مع الصوت نقوم نحن بكتابته على الجدار بسرعة حتى نمثل ما قاله كما هو واضح في الصورة:



أولا سنختار الخلفية المناسبة كما تعلمنا سابقا. بعدها سنبدأ بإختيار الكائنات.



إختر كائن القلم من قائمة الكائنات المتوفرة بالبرنامج



أيضا قم بإختيار الكائن التالي





1 إضغط على كلمة custom من هذا الشريط

2 إضغط هنا

3 قم بتحديد منتصف الصورة على رأس القلم. بحيث هنا يكون منتصف الدوران أيضا

1 إضغط على كلمة custom من هذا الشريط

2 قم بتحميل الأرقام من 2 إلى 5 كما هو واضح

هنا سنعمل على الأرقام من 1 إلى 5 وتكون الأرقام كمظاهر وليست ككائنات منفصلة إنما كائن واحد فقط ب خمسة مظاهر.





1

الآن سنقوم بعملية تسجيل صوت الأرقام حتى يظهر الرقم ونسمع صوته في نفس الوقت.

3

الآن قم بإضافة مقطع جديد للتسجيل

1 إضغط على كلمة sound من هذا الشريط


2 إمسح أول تسجيل. لن نحتاجه هنا

4 قم بالتسجيل عن طريق الضغط على هذا الزر ثم التحدث ثم الضغط عليه عند الإنتهاء

قم بتسجيل إسم الرقم، إبدأ بالرقم 1 حتى الرقم 5 بحيث تنتهي بخمسة تسجيلات



صفحة الأوامر ستكون كالتالي:



أمر التكرار المستمر



نضيف أمر تغيير المظاهر



نضيف أمر تشغيل الصوت






pick random 1 to 5

1 اختر operators من قائمة الأوامر الأساسية



2 قم بنقل pick random إلى نافذة الأوامر

3 حدد الرقمين 1 و 5 بداخل الأمر والذي يعطي رقم عشوائي بين 5 و 1




4 أدخل الأمر هنا



5 سنضيف أمر انتظار 5 ثواني بين الرقم والرقم الآخر

6 نحدد نوع الصوت حسب رقم المظهر فمع الرقم واحد نشغل الصوت "رقم واحد"





الآن سنقوم ببرمجة أوامر الكتابة باستخدام كائن القلم .

```
when clicked  
clear
```

يتم حذف أي رسم سابق مع بداية البرنامج

```
when clicked  
clear  
go to x: 146 y: 40
```

ثم نحدد موقع الكائن (القلم)

```
when clicked  
clear  
go to x: 146 y: 40  
set pen color to 0  
set pen size to 3
```

بعدها نحدد لون الكائن (أبيض) وحجمه (3)

الخطوة التالية هي كيفية كتابة الأرقام بعد سماعها وذلك عن طريق الضغط على الأرقام 1 و 2 و 3 و 4 و 5 من لوحة المفاتيح.

1 أختار الأمر: "عند الضغط على الزر"

when space key pressed

2 عند الضغط على قائمة الخيارات سيظهر التالي

3 إختار الرقم 1 من القائمة

```
when 1 key pressed  
go to x: 146 y: 40  
clear  
pen down  
glide 0.5 secs to x: 146 y: -40  
pen up
```


بعد إختيار الرقم 1 الآن نضع الأوامر التالية حسب ما في الصورة. نحدد مكان الكائن (القلم) ثم نحذف أي كتابة سابقة ثم نضع القلم ونحركه لمدة نصف ثانية بشكل مستقيم ثم نرفع القلم عن الكتابة.





أمر آخر للرقم 1، نعمل بنفس
الفكرة على الرقم 2 عند الضغط عليه. كل
رقم موضح في الصورة التالية يبين الخط
المرسوم.


```
when 2 key pressed
go to x: 146 y: 40
clear
pen down
glide 0.5 secs to x: 186 y: 40 1
glide 0.5 secs to x: 186 y: 0 2
glide 0.5 secs to x: 146 y: 0 3
glide 0.5 secs to x: 146 y: -40 4
glide 0.52 secs to x: 186 y: -40 5
pen up
```



ملاحظة : الرسم الظاهر هو ما ستلاحظه على شاشة العرض .

عند الضغط على الرقم 3 يتم رسم هذا
الرقم. يتم رفع الرقم في حالة الانتقال كما هو
موضح في الخطوة رقم 2.


```
when 3 key pressed
go to x: 146 y: 40
clear
pen down
glide 0.5 secs to x: 186 y: 40 1
glide 0.5 secs to x: 186 y: -40 1
glide 0.5 secs to x: 146 y: -40 1
pen up
go to x: 186 y: 0 2
pen down
glide 0.5 secs to x: 146 y: 0 3
pen up
```






```
when 4 key pressed
go to x: 146 y: 40
clear
pen down
glide 0.5 secs to x: 146 y: 0
glide 0.5 secs to x: 186 y: 0
pen up
go to x: 186 y: 40
pen down
glide 0.5 secs to x: 186 y: -40
pen up
```


نكمل على بقية
الأرقام، الآن نعمل أوامر
رسم الرقم 4 كما هو واضح في
الصورة



```
when 5 key pressed
go to x: 186 y: 40
clear
pen down
glide 0.5 secs to x: 146 y: 40
glide 0.5 secs to x: 146 y: 0
glide 0.5 secs to x: 186 y: 0
glide 0.5 secs to x: 186 y: -40
glide 0.5 secs to x: 146 y: -40
pen up
```

أخيراً ، نضع أوامر رسم الرقم 5.
كما هو واضح في الصورة





```

when clicked
clear
go to x: 146 y: 40
set pen color to 0
set pen size to 3

```

```

when 1 key pressed
go to x: 146 y: 40
clear
pen down
glide 0.5 secs to x: 146 y: -40
pen up

```

```

when 2 key pressed
go to x: 146 y: 40
clear
pen down
glide 0.5 secs to x: 186 y: 40
glide 0.5 secs to x: 186 y: 0
glide 0.5 secs to x: 146 y: 0
glide 0.5 secs to x: 146 y: -40
glide 0.52 secs to x: 186 y: -40
pen up

```

```

when 3 key pressed
go to x: 146 y: 40
clear
pen down
glide 0.5 secs to x: 186 y: 40
glide 0.5 secs to x: 186 y: -40
glide 0.5 secs to x: 146 y: -40
pen up
go to x: 186 y: 0
pen down
glide 0.5 secs to x: 146 y: 0
pen up

```

```

when 4 key pressed
go to x: 146 y: 40
clear
pen down
glide 0.5 secs to x: 146 y: 0
glide 0.5 secs to x: 186 y: 0
pen up
go to x: 186 y: 40
pen down
glide 0.5 secs to x: 186 y: -40
pen up

```

```

when 5 key pressed
go to x: 186 y: 40
clear
pen down
glide 0.5 secs to x: 146 y: 40
glide 0.5 secs to x: 146 y: 0
glide 0.5 secs to x: 186 y: 0
glide 0.5 secs to x: 186 y: -40
glide 0.5 secs to x: 146 y: -40
pen up

```

x: 0
y: 0

في النهاية، ستكون نافذة كتابة الأوامر بهذا الشكل النهائي. يوجد أمر لكل رقم

الآن قم بعمل باقي الأرقام واستمتع بالتحدي مع زملائك

تحدي





حياة المدينة

في هذا التحدي سنتعلم أوامر جديدة تساعد المتحدي على العبور في المدينة بدون أي مشاكل ، حتى يستطيع اجتياز تحديات المدينة.




يوجد نوع من أوامر البرمجة يسمى بالجملة الشرطية ، ويعتبر أحد أهم عناصر البرمجة. يقصد بها أنه لا يحدث العمل إلا بعد عمل شرط معين.

مثال: الهاتف

الهاتف لا يتوقف عن الرنين إلا إذا أجبنا على الإتصال




هذا هو أحد أوامر
الجملة الشرطية




1 إذا كانت الإجابة نعم
(تم الرد على الهاتف)

2 أوقف العمل (الندمة
في هذه الحالة)





الآن لنرجع لتحدي حياة المدينة:



قم باختيار الكائنات الواضحة في الصورة، أيضا قم باختيار الخلفية.

قم بتوزيع الكائنات كما في الصورة التالية، لا تهتم باتجاه الكائنات:



لنبدأ الآن ببرمجة الكائنات:



يتحرك هذا الطفل بمقدار 3 خطوات واذا لمس نهاية الإطار يرجع وينتظر مقدار 0.1 ثانية بين المظهر والآخر





when  clicked

forever

move 7 steps

if on edge, bounce



when  clicked

forever

move 8 steps

if on edge, bounce




لنتقل الآن إلى أوامر القطعة

when  key pressed

repeat 4

point in direction 0

move 10 steps



عند الضغط على زر السهم العلوي في لوحة المفاتيح



تكرار محدد لمدة 4 مرات



إتجاه إلى الأعلى (زاوية صفر)




التحرك بمقدار 10 خطوات

when  key pressed

repeat 4

point in direction 180

move 10 steps



عند الضغط على زر السهم السفلي في لوحة المفاتيح



تكرار محدد لمدة 4 مرات



إتجاه إلى الأسفل (زاوية 180°)




التحرك بمقدار 10 خطوات





هذه المرة سنستخدم
المتحسسات من الأوامر
الرئيسية



إذا لمس الكائن (القطعة)
السيارة الأولى

تصبح القطعة بصوت
(mewo)

ترجع إلى نقطة
البداية

وتفكر بكلمة (Meow)
لمدة ثانية واحدة





الآن يمكننا إضافة الأوامر السابقة على إشارة البداية مع أمر التكرار المستمر كما هو واضح في الصورة التالية.

when  clicked

forever

if **touching** Convertible1 ? then
play sound meow
go to x: 73 y: -142
think Meow for 1 secs

1

2

3

إضغط هنا بزر الفأرة الأيمن وقم بنسخ هذا الأمر كاملا ثم ألصقه أسفل منه. ثم قم بتغيير السيارة في الجملة الشرطية إلى السيارة الأخرى

وتضيف نفس الخطوة للطفل الذي يمشي

when  clicked

forever

if **touching** Convertible1 ? then
play sound meow
go to x: 73 y: -142
think Meow for 1 secs

if **touching** Convertible3 ? then
play sound meow
go to x: 73 y: -142
think Meow for 1 secs


if **touching** Boy3 Walking ? then
play sound meow
go to x: 73 y: -142
think Meow for 1 secs

الجملة الشرطية
للسيارة الأولى



الجملة الشرطية
للسيارة الثانية

الجملة الشرطية
للطفل المار





واصل الطريق أيها
المبرمج الرائع





زيادة الإثارة والتحدي في هذه المرحلة سنضيف مسألة النقاط.
كيف نستخدم النقاط والمتغيرات.



علامة الصح تعني
عرض المتغير على
الشاشة

إخفاء المتغير المحدد

إظهار المتغير المحدد

تغير المتغير بمقدار (1)

تثبيت المتغير على عدد
محدد

وضع المتغير في الوقت
الحالي (كم قيمته الآن)





```
when clicked
  set point to 20
  forever
    if touching Convertible1 ? then
      play sound meow
      go to x: 73 y: -142
      think Meow for 1 secs
      change point by -1
    if touching Convertible3 ? then
      play sound meow
      go to x: 73 y: -142
      think Meow for 1 secs
      change point by -1
    if touching Boy3 Walking ? then
      play sound meow
      go to x: 73 y: -142
      think Meow for 1 secs
      change point by -1
    if 0 = point then
      say Game Over for 2 secs
      stop all
```

سيبدأ التحدي بقوة 20 نقطة لدى المتحدي

تناقص نقطة إذا لمس السيارة الأولى

تناقص نقطة إذا لمس السيارة الأخرى

تناقص نقطة إذا لمس الطفل الذي يمشي

جملة شرطية، في حال النقاط = صفر ، يكتب القم متحدثا (Game Over) لمدة ثابنتين ثم يتوقف التحدي.


```
if 0 = point then
  say Game Over for 2 secs
  stop all
```





تحدي الملاعب/ لعبة التنس

في هذه المرحلة سنواجه لاعب تنس محترف، إنه الحاسوب.
لننشئ ملعب التنس ولنقم بمواجهة الخصم.
علما بأن لدى كل لاعب 5 نقاط.



هنا سنعمل على تصميم مسابقة لعبة التنس من البداية ومكوناتها كالتالي:

*اللاعب الأيمن يتحكم به عن طريق حركة مؤشر الفأرة

*اللاعب الأيسر هو المتحدي (الحاسب الآلي)

*الكرة

*نرسم كائنين جديد بإسم Sprite1 و Sprite 2

*نقوم برسم الخلفية






في هذه الخطوات سنقوم بقلب صورة المؤشر 90 درجة ، أتبع الخطوات التالية :

1





2




3

قم بتحديد الكائن من هذه القائمة





قم بعمل نفس العملية مع الكائن الآخر





1 أوامر اللاعب

1

ثبتنا الإتجاه الأفقي حتى لا يتقدم اللاعب عن خط محدد كما أعطيناها أمر الإتجاه لمؤشر الفأرة والحركة 6 حركات كل مرة

```
when clicked
forever
  set x to 220
  point towards mouse-pointer
  move 6 steps
```

2 أوامر المتحدي

2

ثبتنا الإتجاه الأفقي حتى لا يتقدم المتحدي عن خط معين كما أعطيناها أمر الإتجاه بإتجاه الكرة والحركة 12 حركة كل مرة.

```
when clicked
forever
  set x to -220
  point towards Basketball
  move 6 steps
```

3 أوامر للكرة

3

عمل أمر حساب النقاط (5 نقاط) عند بدء اللعبة، ثم حددنا مكان بداية الحركة ويكون إتجاه الكرة بشكل عشوائي من 0 إلى 360

```
when clicked
set points to 5
go to x: 0 y: 0
point in direction pick random 0 to 360
```

لعمل النقاط إذهب للأوامر الرئيسية Data. أختَر Make a variable. ثم اكتب اسم الخاصية وبعدها اسحبها في نافذة الأوامر.

نستخدم هذا الأمر للاختيار رقم عشوائي من بين الأرقام المحددة. نجد هذا الأمر في نافذة الأمر الرئيسي operators





أوامر للكرة

4

```
when green flag clicked
  forever loop
    move 12 steps
    if on edge, bounce
    if touching Button2 ? then
      point in direction pick random 200 to 330
    if touching Button3 ? then
      point in direction pick random 30 to 150
    if touching Sprite1 ? then
      change points by -1
    if touching Sprite2 ? then
      change points by 1
```

تكملة في أوامر الكرة، عمل أمر حركة للكرة بمقدار 12 وعند الإصطدام بالحائط ترتد. إذا لمس اللاعب الأيمن الكرة ترتد بزاوية عشوائية وكذلك إذا حدث مع المتحدي. مجال الزاوية تم تحديده حتى تتجه بإتجاه الشخص الآخر. إذا لمست العمود خلف اللاعب فإن النقاط تتناقص بمقدار نقطة وإذا اصطدمت الكرة بالعمود خلف المتحدي تزيد بمقدار نقطة.





الآن علينا رسم الخلفية حتى تعطينا النتيجة النهائية.
الأعمدة الخلفية التي قمنا برسمها من قبل لا يكون بها أي برمجة.



قم بإضافة الأمر التالي: عند
وصول النقاط إلى صفر أو
عشرة أوقف اللعبة

تحدي





تحدي الأصوات

لعبة الأصوات هي تحدي رائع يعرفنا كيف نستخدم اصواتنا كمتحسسات في الألعاب، سيتحرك جوبو إلى يمين ويسار وعند سماعه لصوت مرتفع سوف يقفز للأعلى للإلتقاط الكرة: لنبدأ في تصميم هذه اللعبة بمعرفة متطلبات اللعبة



*أنزل صورة Gobo من الكائنات المتوفرة



*قم بإختيار كرة أيضا من الكائنات

*أختر خلفية حسب رغبتك.





1



نبدأ بأوامر Gobo


نعمل على أمر القفز بحيث نستخدم أمر الإنزلاق من أوامر الحركة motion بحيث عندما يقفز يتغير الإتجاه العمودي ويثبت الأفقي.



4

برمجة جوبو، نحدد موقع الإنطلاق ثم بإدخال تكراراً مستمر تحريكه بمقدار 3 خطوات مع اضافة امر أرتداد عن وصوله للحد.

إذا كان شدة الصوت من الخارج أكثر من 10 فإنه سيحقق شرط الدخول والذي سينفذ الأوامر التالية، قم بالإنزلاق إلى موقع X الحالي وغير العمود إلى 140 ثم انتظر 0.1 ثانية ثم عد إلى موقعك الأولي. وكل هذه الأوامر بداخل التكرار المستمر.



خاصية loudness تجدها ضمن الأوامر الأساسية sensing ويتم ادخالها في أوامر الأمر الرئيسي Operator وفي هذه اللعبة الأمر (>) هو المستخدم.





سنبدأ التحكم في الكرة والتي ستظهر في مكان عشوائي وعندما يلتقطها جوبو تختفي وتضاف لنا نقطة ثم تختفي الكرة وتظهر بعد مدة من الزمن

أوامر الكرة

2

```
when clicked
  set points to 0
  go to x: pick random -240 to 240 y: 150
  show
  forever
    change color effect by 10
    if touching Gobo ? then
      hide
      change points by 1
      wait 5 secs
      go to x: pick random -240 to 240 y: 150
      show
```




أولاً، نستخدم النقاط والتي تكون صفراً ثم نحدد موقع معين لظهور الكرة فيها وبعدها نختار وضعية الظهور، داخل تكرار مستمر نضع تغيير لوني. وبداخل جملة شرطية نتحقق إذا لمس جوبو الكرة. فإن الكرة تختفي ثم نحقق نقطة وننتظر 5 ثواني بعدها يحدد مكان عشوائي آخر للكرة ثم تظهر بذلك المكان.





عن طريق نسخ ولصق الكرة في نافذة الكائنات نستطيع عمل كرة إضافية تعمل نفس عمل الكرة الأولى بحيث تضيف التحدي في اللعبة، يمكنك مضاعفة الكائن عن طريق الضغط عليه بالزر الأيمن ثم الضغط على كلمة duplicate



قم بتحريك الكرات أثناء اللعب
وظهورها وإختفائها تلقائيا قبل
إلتقاطها.

تحدي





تحدي المزرعة

في هذا المثال سنقوم بتجميع التفاح المتساقط من الأعلى في داخل السلة. في حال تجميعنا تفاحة واحدة نحصل على عشر نقاط.



ماذا نحتاج في هذا التحدي:



*تفاحة، من نافذة الكائنات الجديدة



*سلة ، نقوم برسمها بأنفسنا (نضع خط أصفر بالداخل)

*خلفية ، أختار الخلفية المناسبة





أوامر السلة 1

عند الضغط على السهم الأيمن المتواجد في لوحة المفاتيح تتحرك السلة بمقدار 10 خطوات ويتكرر هذا الأمر 10 مرات.

```
when right arrow key pressed
repeat 10
  move 10 steps
```

أوامر السلة 2

عند الضغط على السهم الأيسر المتواجد في لوحة المفاتيح تتحرك السلة بمقدار -10 خطوات (باتجاه الأيسر) يتكرر 10 مرات.


```
when left arrow key pressed
repeat 10
  move -10 steps
```

أوامر التفاحة 3

في برمجة التفاحة نضع أمر البيانات، وبداخل تكرار مستمر نضع موقع أفقي عشوائي للتفاحة مع تثبيت الموقع العمودي. إظهار التفاحة في حالة الإخفاء وبداخل تكرار محدد لـ 50 مرة تتحرك التفاحة بشكل عمودي 10 خطوات. بداخل جملة شرطية إذا لمست التفاحة اللون الأصفر (اللون بداخل السلة) نحصل على عشرة نقاط ثم يظهر خبر حصولنا على النقاط وبعدها تختفي التفاحة.


```
when clicked
set scr to 0
forever
  go to x: pick random -200 to 300 y: 179
  show
  repeat 50
    move 10 steps
    if touching color yellow ? then
      change scr by 10
      think 10 for 1 secs
      hide
```





قم بقلب زاوية كائن التفاحة
إلى 180 درجة ،
وذلك لتحديد حركة الكائن للأسفل

هذا هو الشكل النهائي للتحدي ومن خلاله تستطيع الإستمتاع بتجميع التفاح
بشكل بسيط



قم بإضافة قبلة، يخسر
اللاعب 5 نقاط عند إتقاطها !

تحدي

قم بإضافة وقت لتجميع التفاح،
أيضا أضف تفاحة باللون الأخضر
يكسب اللاعب 5 نقاط فقط من
إتقاطها.

تحدي





العلم والأدب .. خير من كنوز الفضة والذهب








تحدي الفضاء

التحدي القادم يخرجنا إلى الفضاء الخارجي إلى حرب الفضاء. في هذا التحدي سنستخدم الطائرة لهزيمة سفن فضاء العدو. نكسب نقطة عند تدمير كل سفينة فضاء. معنا 5 فرص لإصطدام سفن العدو بنا



- ماذا نحتاج من كائنات:
- *قم برسم طائرة خاصة بك 
 - *قم برسم سفينة فضاء للعدو 
 - *قم برسم ذخيرة 
 - *إختر خلفية مناسبة





أضف مظهر آخر لكل من الكائنات التالية:



2 أمر تحرك الطائرة للأعلى

```
when up arrow key pressed
point in direction 0
move 3 steps
```

1 أمر تحرك الطائرة لليمين

```
when right arrow key pressed
point in direction 90
move 5 steps
```

4 أمر تحرك الطائرة للأسفل

```
when down arrow key pressed
point in direction 180
move 3 steps
```

3 أمر تحرك الطائرة لليساار

```
when left arrow key pressed
point in direction -90
move 5 steps
```

5 أمر إطلاق النار

```
when space key pressed
switch costume to costume2
wait 0.2 secs
switch costume to costume1
broadcast shot
```

عند الضغط على زر المسافة فإن مظهر الطائرة يتغير إلى المظهر الثاني لمدة 0.2 ثانية ثم يعود إلى المظهر الأول وبعدها يرسل رسالة باسم shot يمكن تغيير اسم الرسالة من القائمة المنسدلة وإختيار اسم جديد new message





6 أمر الطائرة

عند البدء، نعمل أوامر البيانات، نقاط points و حياة lives ، النقاط نحصل عليها عند ضرب الصحن الفضائي و الحياة نخسرها عندما نصطدم بالعدو. نضع أمر تحديد شكل البداية للطائرة، ثم نحدد موقع البداية ونظهر في حالة الإختفاء.

```
when clicked
set points to 0
set Lives to 5
switch costume to costume1
go to x: 0 y: -138
show
```

7 أمر الطائرة

نكمل مع أوامر الطائرة بعمل تكرار مستمر للجملة الشرطية، في حالة كان عدد الحياة أقل من 1 فإن أمر التكرار المحدد يعمل على تشغيل مؤثر مثل الانفجار للطائرة. ثم التبديل بين المظهر الثالث والرابع وبعدها تتوقف اللعبة

```
when clicked
forever
if Lives < 1 then
repeat 5
switch costume to costume3
wait 0.1 secs
switch costume to costume4
wait 0.1 secs
stop all
```

1 أمر صحن الفضاء

عند بدء الأوامر، يتحدد مظهر صحن الفضاء الأول ويكون موقعه العمودي محدد ولكن الأفقي عشوائي

```
when clicked
switch costume to costume1
go to x: pick random -220 to 220 y: 160
```





2 أمر صحن الفضاء

نكمل في نفس الأوامر، بداخل أمر تكرار مستمر نظهر الكائن ثم نعمل تغيير جرافيكلي وتحرك بمقدار 2 خطوة.

```
forever
  show
  change color effect by 25
  move 2 steps
```

3 أمر صحن الفضاء

نضع الأوامر التالية داخل التكرار المستمر، في حالة لمس الكائن لحد الشاشة فإنه يختفي ثم بعد ثانية يتحدد مكان جديد له ويظهر من جديد

```
if touching edge ? then
  hide
  wait 1 secs
  go to x: pick random -220 to 220 y: 160
  show
```

4 أمر صحن الفضاء

إذا لمست الرصاصة الكائن، يتوقف التأثير الجرافيكلي ويتحول إلى المظهر الثاني ينتظر بعض الوقت ثم تزيد النقاط بمقدار 1 بعدها ينتظر ثانية واحدة ويعيد ترتيب مكانه والرجوع إلى المظهر الأولي.

```
if touching Beetle ? then
  clear graphic effects
  switch costume to costume2
  wait 0.2 secs
  hide
  change points by 1
  wait 1 secs
  go to x: pick random -220 to 220 y: 160
  switch costume to costume1
  show
```





5 أمر صحن الفضاء

أيضاً بأسفل الأوامر السابقة، إذا لمس الكائن الطائرة فإن الكائن يختفي ثم تنقص الحياة بمقدار 1 بعدها ينتظر ثانية ويظهر في مكان عشوائي على نفس الارتفاع.

```
if touching Airforce ? then
  hide
  change Lives by -1
  wait 1 secs
  go to x: pick random -220 to 220 y: 160
  show
```

ملاحظة

الأوامر 5 4 3 2

هي أوامر بداخل أمر تكرار مستمر واحد

وملتصقه بالأمر 1

6 أمر صحن الفضاء

في حالة حصول اللاعب على أكثر من 30 نقطة فإن صحن الفضاء ينسخ نفسه كل 10 ثواني.

```
when clicked
  forever
    if points > 30 then
      create clone of myself
    wait 10 secs
```





7 أمر صحن الفضاء

عندما ينسخ الكائن نفسه فإننا يجب أن نعمل له أوامر خاصة. نلاحظ تشابه الكائن النسخة في الأوامر مع الكائن الأصلي. في الجملة الشرطية بدل الاختفاء نعمل حذف الكائن النسخة لأننا لا نريده أن يرجع وإنما أن ينتهي. وذلك لأنه توجد نسخة جديدة كل 10 ثواني

```
when I start as a clone
switch costume to costume1
go to x: pick random -220 to 220 y: 160
forever
show
change color effect by 25
move 2 steps
if touching edge ? then
delete this clone
if touching bullet ? then
clear graphic effects
switch costume to costume2
wait 0.2 secs
change points by 1
delete this clone
if touching Airforce ? then
change Lives by -1
delete this clone
```

يمكنك بعد إنتهاء الأوامر نسخ الكائن ليظهر لك كائن آخر بنفس الأوامر ، قم بتبديل القليل في الرسم الجديد واستبدل ما تريد حسب رغبتك





أمر الرمية

1

أمر إخفاء الكائن

when clicked

hide

أمر الرمية

2

أمر نسخ الكائن لنفسه

when I receive shot

create clone of myself

أمر الرمية

3

بعد نسخ نفسه ، تذهب النسخة إلى الطائرة بعدها تتقدم بمقدار 40 خطوة لتتواجد في مقدمة الطائرة ثم يظهر الكائن وينطلق في ثانيتين من موقع الطائرة الأفقي الحالي إلى نهاية الشاشة ثم تختفي

when I start as a clone

go to Airforce


change y by 40

show

glide 2 secs to x: x position y: 180


delete this clone





قم بقلب زاوية كائن سفينة
فضاء العدو إلى 180 درجة ،
وذلك لتحديد حركة الكائن للأسفل

الشكل النهائي لهذه المرحلة من التحدي كالتالي:




أضف خيارات متعددة مثل الوقت
الزمني، نسخ الصحن الطائرة نفسه
، إختلاف نوع الرصاص عند عدد
معين من النقاط.

تحدي





في هذا التحدي سيمر المتحدي حول المتاهة ويجتاز العقبات التي أمامه حتى يصل إلى نقطة الكنز ليفوز بالتحدي، علما بأن المتحدي له فقط 5 فرص للنجاح.



الكائنات المطلوبة:



1-المتسابق ، قم بأختياره، في المثال هو



2-الصواعق

3-لمروحة ، قم برسمها حسب رغبتك




4-كرة



5-مفتاح





6. صخرة 

7. نجمة 



8. شبح 

9. علامة الكنز 

10. الجدار، قم برسمه بنفسك

*ثم اختر خلفية مناسبة

سيتم التعديل على المظاهر حسب الشكل التالي:



ثم قم بإعادة ترتيب الأشكال بحيث ترفع الأخير ليكون الأول ، والثالث ليكون الثاني و الثاني ليكون هو الثالث والأول هو الرابع. كما هو واضح في الصورة التالية.





1

نقوم بعمل الإتجاهات
للتحكم بالمتحدي كما
تعلمنا فيما سبق لأربع
إتجاهات.

```
when up arrow key pressed
point in direction 0
move 10 steps

when right arrow key pressed
point in direction 90
move 10 steps

when down arrow key pressed
point in direction 180
move 10 steps

when left arrow key pressed
point in direction -90
move 10 steps
```

2

نبدأ بإختيار الخلفية وذلك في حال تغييرها. تبدأ
بالخلفية التي اخترتها من القائمة. بعدها نضيف
أمر المتغيرات ونحدد بدء التحدي ب 5 فرص
ونحدد موقع الكائن وإتجاهه. بعدها تكرر محدد
3 مرات ليعطي إشارة البداية


```
when clicked
switch backdrop to rays
set Lives to 5
go to x: -209 y: -143
point in direction 90
repeat 3
show
wait 0.1 secs
hide
wait 0.1 secs
show
```





```
when clicked
  forever
    if touching wall ? then
      move 10 steps
    if touching Button5 ? then
      switch backdrop to Win
      stop all
    if Lives < 1 then
      switch backdrop to Lose
      stop all
```

3
بداخل تكرار مستمر إذا لمس الجدار يرجع 10 خطوات وإذا لمس رمز الكنز Button5 يتحول إلى خلفية الفوز ثم يوقف اللعبة. وإذا كان عدد الحياة أقل من 1 فإن الخلفية تتحول إلى خلفية الخسارة ثم تتوقف اللعبة.



لا تنسى أن تقوم بإنشاء خلفية للفوز وأيضاً للخسارة

ملاحظة : يتم تحديد اسم الخلفية حسب ما اسميتها عند الإنشاء.
أخر من القائمة الاسم الذي حددته لكل خلفية.



4
إذا لمس الكائن البرق الاول أو البرق الثاني (سنعمله فيما بعد) أو المروحة أو الكرة أو الشبح فإنه يرجع مكانه وتنقص نقاط الحياة بمقدار 1

```
when clicked
  forever
    if touching Lightning ? or touching Fan ? or touching Lightning2 ? or touching Beachball ? or touching Ghoul ? then
      go to x: -209 y: -143
      change Lives by -1
```





```
when clicked
  wait 0.6 secs
  switch costume to lightning1
  forever
    show
    repeat 4
      next costume
      wait 0.15 secs
    hide
    wait 1 secs
```



يظهر تأثير البرق كأحد التحديات في هذه اللعبة. يبدأ بالظهور بالشكل الأولي ثم بداخل تكرار مستمر يتغير المظهر أربع مرات مروراً بكل المظاهر الأربعة. ثم تختفي لمدة ثانية.

قم بنسخ الكائن ووضعه على مسافة من الكائن السابق بحيث تكون تحديات مستمرة وغير التوقيت في أول الأمر.

```
when clicked
  go to x: -210 y: 151
  show
  forever
    change color effect by 25
    if touching beetle ? then
      clear graphic effects
      go to front
      go to beetle
    if touching Rocks ? then
      wait 0.2 secs
      hide
```



أمر تحديد موقع الكائن ثم يأمر بالظهور، بداخل أمر تكرار مستمر، أمر تغيير ألوان الكائن. إذا لمس المتحدي المفتاح امسح التأثير الجرافيكى ثم إذهب للمقدمة (حتى يكون في ظهر المتحدي) بعدها يحدد موقعه بموقع المتحدي.

إذا لمس المفتاح الصخرة فإنه ينتظر قليلاً ثم يختفي





```
when green flag clicked
show
forever
  next costume
  wait 0.2 secs
  change color effect by 5
  turn 15 degrees
  if touching beetle ? then
    hide
    broadcast lastlevel
```



ظهور الكائن ثم في أمر تكرار مستمر ظهور المظهر التالي للكائن مع تغيير الألوان ودوران الكائن 15 درجة. إذا لمس المتحدي النجمة تختفي ويرسل رسالة باسم النقطة النهائية.

```
when green flag clicked
set size to 55 %
go to x: -21 y: -21
show
forever
  if touching key ? then
    repeat 13
      move -5 steps
    repeat 5
      show
      wait 0.1 secs
      hide
      wait 0.1 secs
```



تحديد حجم الكائن ثم يتم تحديد المكان للكائن وظهوره. بداخل التكرار المستمر إذا لمس المفتاح يتحرك للخلف ثم يبدأ بالظهور والاختفاء وبعدها يختفي.





when  clicked

go to x: -153 y: 157

show

forever

glide 2 secs to x: -145 y: 2

glide 3 secs to x: 211 y: 4

point in direction -90

glide 3 secs to x: -134 y: 4

glide 2 secs to x: -139 y: 150

point in direction 90

if  touching  ? then

switch costume to 

wait 0.3 secs

switch costume to 

wait 0.2 secs

hide

go to x: -153 y: 157



تحديد مكان الشبح ثم نظهره.
في أمر تكرار مستمر نعمل على
تحريك الشبح للأسفل ثم لليمين
بعدها يعكس اتجاهه إلى
الشمال ويرجع أدراجه ثم يصعد
ويعيد اتجاهه. إذا لمس المتحدي
فإنه يتغير إلى المظهر التالي
ويرجع للمظهر الأول ثم يختفي
ويرجع لموقعه.

when I receive 

switch costume to 

wait 0.3 secs

switch costume to 

wait 0.2 secs

hide

إذا استلم رسالة النقطة النهائية فإنه يظهر
المظهر التالي ثم يظهر المظهر الاول بعدها
يختفي





```
when clicked
go to x: -216 y: 169
show
forever
  move 2 steps
  show
  if touching wall ? then
    hide
    wait 3.5 secs
    go to x: -216 y: 169
  if touching beetle ? then
    hide
    wait 3.5 secs
    go to x: -216 y: 169
```



أمر الكرة بأن تتواجد في المكان المحدد ثم تظهر. في أمر تكرار مستمر تتحرك الكرة خطوتين ثم تظهر في حالة الإختفاء، إذا لمس الكائن الجدار يختفي ثم ينتظر 3.5 ثانية ثم يرجع لموقعه. إذا لمس الكائن المتحدي يختفي أيضا و يتواجد في موقعه الأولي.

```
when I receive lastlevel
forever
  next costume
  wait 0.2 secs
```



إذا استلم رسالة النقطة النهائية فإن الكائن بشكل مستمر يظهر الكائن التالي.

```
when clicked
forever
  turn 15 degrees
  wait 1 secs
```



برمجة الطاحونة بحيث تتحرك بشكل دوراني دائما.



تحدي الصحراء


في هذا التحدي سنحتمي بداخل خيمة في الصحراء من الدواب والحشرات. سيكون معنا سلاح يتحرك مع مؤشر الفأرة والذي يطلق الرصاص بمجرد الضغط على زر المسافة. لديك عشرة دقائق لإنهاء المهمة.



قم بتنزيل الكائنات التالية من الكائنات المتوفرة في البرنامج :

1- الخيمة والرصاص هي كائنات تم رسمها  

2- يمكنك استخدام خلفية مناسبة أو رسم أخرى كما فعلنا هنا.

3- استخدمنا العصا السحري و عدلنا به ليكون سلاح ناري. 

4- الفأر . 

5- الخنفساء . 






```
when clicked
go to x: 0 y: 0
forever
  point towards mouse-pointer
```



بداخل أمر متكرر إجعل اتجاه العصا مع اتجاه مؤشر الفأرة، بهذا الأمر ستتحرك العصا عند تحريك مؤشر الفأرة وفي إتجاهها.



```
when clicked
hide

when space key pressed
go to x: 0 y: 0
show
point towards mouse-pointer
repeat until touching edge ?
  move 10 steps
hide
```



إخفاء الرصاصة عند التشغيل



في حالة الضغط على زر المسافة يتمركز الكائن في منتصف الصفحة ثم يظهر بعدها يتجه بإتجاه مؤشر الفأرة، يتحرك الكائن حتى يلامس حدود العرض ثم يختفي (بداخل أمر شرطي تكرار لحد....)





```
when clicked
clear
go to x: 100 y: 180
show
forever
  move 1 steps
  point towards tent
  if touching bullet ? then
    change Points by 1
    stamp
    hide
    set x to pick random -240 to 240
    wait 5 secs
    show
    go to x: x position y: 180
  if touching tent ? then
    wait 0.1 secs
    hide
    wait 5 secs
    show
```

تصفية الشاشة في بداية الأمر ثم نحدد موقع الإنطلاق بعدها إظهار الكائن، بداخل تكرار مستمر تحرك بمقدار خطوة والإتجاه إلى كائن الخيمة ، بداخل أمر شرطي، إذا لمست الرصاصة الكائن فإن النقاط تزيد بمقدار واحد ويطبع الكائن نفسه في العرض (عمل نسخة مطبوعة منه) ثم يختفي ويتغير مكان الإتجاه الأفقي وينتظر 5 ثواني ثم يظهر مرة أخرى ثم يذهب للإحداثيات الجديدة. إذا لمس كائن الخيمة ينتظر 0.1 ثانية حتى نتأكد من لمس الخيمة بعدها يختفي ثم ينتظر 5 ثواني ليظهر مرة أخرى

```
when clicked
hide
wait 3 secs
go to x: 222 y: -50
show
```

بعد الإنتهاء من هذا الأمر يمكننا عمل نسختين إضافيتين من هذا الكائن مع إضافة تغيير بسيط في مقدمة البرنامج، كالتالي بحيث يظهر بتوقيت مختلف، نعطيه لواحد منهم 3 ثواني والأخر 4.5 ثانية.





```
when clicked
hide
wait 1 secs
go to x: -219 y: -114
show
forever
  move 0.75 steps
  point towards tent
  if touching bullet ? then
    change Points by 1
    stamp
    hide
    set y to pick random -180 to 180
    wait 5 secs
    show
    go to x: -240 y: y position
  if touching tent ? then
    wait 0.1 secs
    hide
    wait 5 secs
    show
```

أمر كائن الفأرة مشابه جداً لأمر الحشرة ولكن يختلف بشكل بسيط في الإتجاهات والتوقيت. لاحظ الفرق وقم بتغيير ما يناسبك



قم بعمل نفس النسخ كما عملنا سابقا مع تغيير بعض التوقيتات والإتجاهات ، لا تجعل الحشرات تكون في جانب والفئران في جانب آخر وإنما ادمج بينهم





جميع أوامر الكائنات الأربعة الإضافية المنسوخة



Ladybug3

```
when clicked
hide
wait 4.5 secs
go to x: 222 y: -50
show
forever
  move 0.75 steps
  point towards tent
  if touching bullet ? then
    change Points by 1
    stamp
    hide
    set y to pick random -180 to 180
    wait 5 secs
    show
    go to x: 240 y: y position
  if touching tent ? then
    wait 0.1 secs
    hide
    wait 5 secs
    show
```




Mouse3

```
when clicked
hide
wait 5 secs
go to x: 100 y: -180
show
forever
  move 1 steps
  point towards tent
  if touching bullet ? then
    change Points by 1
    stamp
    hide
    set x to pick random -240 to 240
    wait 5 secs
    show
    go to x: x position y: -180
  if touching tent ? then
    wait 0.1 secs
    hide
    wait 5 secs
    show
```



باقي الأوامر





جميع أوامر الكائنات الأربعة الإضافية المنسوخة



```
when clicked
hide
wait 5 secs
go to x: 100 y: -180
show
forever
  move 1 steps
  point towards tent
  if touching bullet ? then
    change Points by 1
    stamp
    hide
    set x to pick random -240 to 240
    wait 5 secs
    show
    go to x: x position y: -180
  if touching tent ? then
    wait 0.1 secs
    hide
    wait 5 secs
    show
```



```
when clicked
hide
wait 4.5 secs
go to x: 222 y: -50
show
forever
  move 0.75 steps
  point towards tent
  if touching bullet ? then
    change Points by 1
    stamp
    hide
    set y to pick random -180 to 180
    wait 5 secs
    show
    go to x: 240 y: y position
  if touching tent ? then
    wait 0.1 secs
    hide
    wait 5 secs
    show
```





```
when clicked
go to x: 0 y: 0
set Points to 0
set time to 1000
forever
repeat 1
change time by -1
if touching Ladybug1 ? then
change Points by -5
if touching Ladybug2 ? then
change Points by -5
if touching Ladybug3 ? then
change Points by -5
if touching Mouse1 ? then
change Points by -5
if touching Mouse2 ? then
change Points by -5
if touching Mouse3 ? then
change Points by -5
if Points < 0 then
stop all
if time < 0 then
stop all
```

نبدأ بخطوتين وهي عمل المتغيرات، نقاط ووقت . النقاط تبدأ بالصفير والوقت نضبطه على 1000 إذا أردنا ألف ثانية ، يمكنك جعلها 500 لتكون خمسمائة ثانية


في أمر التكرار المستمر يوجد أمر تكرر محدد نعيد مرة واحدة إنقاص عدد 1 من الأرقام . بداخل جمل شرطية يخسر المتحدي 5 نقاط مع كل مرة تلمس إحدى الكائنات الخيمة وإذا كان عدد النقاط أقل من صفر فإن اللعبة تتوقف وإذا كان الوقت أقل من صفر أيضاً اللعبة تتوقف .

ملاحظة :
تأكد من أن موقع الخيمة يكون في منتصف العرض .





الشكل النهائي لهذه المرحلة من التحدي كالتالي:



قم بعمل إشارات البداية للإنتلاق ،
كعداد البداية ، يمكنك أيضا عمل نسخ
من الرصاصة وليس طلقة واحدة حتى
تختفي.

تحدي





في هذا التحدي سنساعد الكائنات للوصول إلى الأرض، وذلك باستخدام خاصية الكاميرا المتواجدة ببرنامج سكراتش بحيث نحرك الكائنات بأيدينا. لديك 5 كائنات يجب وصولها للموقع الكرة الأرضية ، تفوز باللعبة إذا وصلت جميعا بدون أن تلمسها الحجارة المتساقطة




الكائنات المستخدمة في هذا التحدي هي كما تلاحظ في الصورة والتي تتواجد بنافذة الكائنات .





ملاحظة: يجب أن يكون جهازك مزودا بكاميرا ويب لتتمكن من تطبيق هذه اللعبة !





الكائن الذي سنعمل عليه في البداية هو الكائن الموضح ثم يمكننا نسخ الأوامر مع التعديل على الكائنات التالية التي تعمل نفس العمل

الكائنات التالية هي رسومات تم عملها من نافذة الرسم بالبرنامج



الكائن التالي تم عمل مظاهر له حتى يتحرك عند ملامسته إحدى الكائنات.





```
when clicked
  set size to 45 %
  go to x: -190 y: -114
  repeat 4
    show
    wait 0.2 secs
    hide
    wait 0.2 secs
  show
  glide 1 secs to x: -190 y: 0
  forever
    if video motion on this sprite > 50 then
      repeat 3
        point towards Earth
        move 10 steps
        next costume
        wait 0.2 secs
    if touching Earth ? then
      repeat 10
        change size by -10
        move 1 steps
      change Points by -1
      hide
      broadcast Bat
    if touching Wall ? then
      move -10 steps
```

جميع الأوامر التالية هي أوامر معروفة ما عدا أمر حساسات الكاميرا

أمر الكاميرا يبين بأن تحسس الكاميرا يحدث للكائن عند تجاوز الفرق بمقدار 50

ملاحظة: في بعض الأجهزة ستحتاج إلى تعديل الرقم بحيث يتوافق مع قوة الإضاءة

من الأوامر الرئيسية events نجد خاصية broadcast بحيث نرسل رسالة بإسم نختاره بأنفسنا، والتي تهدف إلى بدء العمل في كائن ثاني، فهي عبارة عن طريقة لربط كائن بكائن أو حدث بحدث





```
when I receive Bat
  wait 1 secs
  set size to 45 %
  go to x: -190 y: -114
  repeat 4
    show
    wait 0.2 secs
    hide
    wait 0.2 secs
  show
  glide 1 secs to x: -190 y: 0
  forever
    if video motion on this sprite > 50 then
      repeat 3
        point towards Earth
        move 10 steps
        next costume
        wait 0.2 secs
    if touching Earth ? then
      repeat 10
        change size by -10
        move 1 steps
      change Points by -1
      hide
      broadcast Beetle
    if touching Wall ? then
      move -10 steps
```

when clicked

hide



بالنسبة للوطواط، فإنه سيكون مخفي لحد حصوله على إشارة البداية من الرسالة التي أرسلناها له في أوامر الكائن السابق. نلاحظ بأن الأوامر متشابهة لذا يمكننا نسخ أوامر الكائن السابق لهذا الكائن وعمل نفس الحركة مع الكائنات القادمة.

نغير الاسم في خاصية broadcast بحيث نرسل رسالة باسم نختاره بأنفسنا، والتي تكون هنا باسم الكائن التالي أو أي اسم مختلف.





when I receive Beetle

wait 1 secs

set size to 45 %

go to x: -190 y: -114

repeat 4

show

wait 0.2 secs

hide

wait 0.2 secs

show

glide 1 secs to x: -190 y: 0

forever

if video motion on this sprite > 50 then

repeat 3

point towards Earth

move 10 steps

next costume

wait 0.2 secs

if touching Earth ? then

repeat 10

change size by -10

move 1 steps

change Points by -1

hide

broadcast Octopus

if touching Wall ? then

move -10 steps


change Live by 1

when clicked

hide

الآن سنعمل على نسخ الأوامر في الكائن السابق ولصقها في الكائنات التالية حسب ما هو واضح





```
when I receive Octopus
```

```
wait 1 secs
```

```
set size to 45 %
```

```
go to x: -190 y: -114
```

```
repeat 4
```

```
show
```

```
wait 0.2 secs
```

```
hide
```

```
wait 0.2 secs
```

```
show
```

```
glide 1 secs to x: -190 y: 0
```

```
forever
```

```
if video motion on this sprite > 50 then
```

```
repeat 3
```

```
point towards Earth
```

```
move 10 steps
```

```
next costume
```

```
wait 0.2 secs
```

```
if touching Earth ? then
```

```
repeat 10
```

```
change size by -10
```

```
move 1 steps
```

```
change Points by -1
```

```
hide
```

```
broadcast Gobo
```

```
if touching Wall ? then
```

```
move -10 steps
```

```
when clicked
```

```
hide
```







```
when I receive Gobo
  wait 1 secs
  set size to 45 %
  go to x: -190 y: -114
  repeat 4
    show
    wait 0.2 secs
    hide
    wait 0.2 secs
  show
  glide 1 secs to x: -190 y: 0
  forever
    if video motion on this sprite > 50 then
      repeat 3
        point towards Earth
        move 10 steps
        next costume
        wait 0.2 secs
    if touching Earth ? then
      repeat 10
        change size by -10
        move 1 steps
      change Points by -1
      hide
    if touching Wall ? then
      move -10 steps
```

```
when clicked
  hide
```





هنا أوامر الصخور المتساقطة والتي تنقص من النقاط المعطاه






هنا أوامر الكرة الأرضية مضافة لها أوامر المتغيرات وأوامر إيقاف البرنامج في حال الخسارة أو الفوز.





```
when clicked
  set Live to 10
  set Points to 5
  turn video on
  forever
    move 2 steps
    if on edge, bounce
    if touching color [black] ? then
      point in direction direction - 180
    if Points < 1 then
      turn video off
      broadcast Win
      stop all
    if Live < 1 then
      turn video off
      broadcast Lost
      stop all
```




لحساب عدد فرص ملامسة الصخور المتساقطة

لحساب عدد الكائنات التي وصلت للأرض








قم بقلب زاوية الصخور المتساقطة
إلى 180 درجة ،
وذلك لتحديد حركة الكائن للأسفل



باقي الكائنات لا تحتاج للبرمجة









هذا مرجع لتتعرف من خلاله على واجهة الرسم وكيفية إستخدام الأدوات الموجودة به

في حال ظهور أكثر من صورة هنا فهذا يدل بأن الرسمة يمكن أن تتحرك أو تستخدم أشكال مختلفة لكائن واحد

عند الضغط على custom تظهر لنا هذه النافذة

زر التوسط، المسؤول عن منتصف الصورة. يحدد نقطة إنتصاف الكائن



زر الإنعكاس من اليمين لليسار

زر الإنعكاس من الأعلى للأسفل






زر تنظيف الشاشة



زر إضافة كائن مع الكائن الحالي من البرنامج


زر إضافة كائن مع الكائن الحالي يتم جلبه من خارج البرنامج




بالضغط على هذا الزر يمكنك الضغط على الرسم في الشكل ثم التعديل فيه




أدوات الرسم الإعتيادية من خط ومستقيم ومستطيل ودائرة وكتابة و صبغ

زر النسخ يقوم بتكرار الكائن في نفس الرسمة من خلال الضغط على الزر ثم الضغط على الكائن






هذه الأزرار وظيفتها ترتيب الأشكال على حسب المستوي الذي تريده



زر التجميع/فصل التجميع هو الزر المسؤول عن تفكيك أجزاء الصورة/زر التجميع/فصل التجميع هو الزر المسؤول عن تفكيك





جميع الخصائص السابقة
تجدها في حال الرسمة
في وضع vector

إضغط على هذا الزر لتحويله إلى bitmap

خاصية vector من الخصائص التي
تسمح بتواجد أكثر من كائن ويمكن
التحكم بهم بشكل منفصل. كما
تتواجد الخصائص السابقة

ملاحظة، في حال تم تحويل الكائن من
خاصية vector إلى bitmap ثم الرجوع
إلى vector يفقد الرسم خاصية التجميع
وفك التجميع.

خاصية bitmap من الخصائص التي لا
تسمح بتواجد أكثر من كائن بحيث اذا
قمت برسم أي شيء جديد يتم
رسمه أعلى الرسم السابق ولا
تستطيع التحكم به بعد الإنتهاء منه



ملحق تعاريف القوانين

الملحق عبارة عن الأوامر مع الشرح ويتضمن مثال مع التوضيح

Motion

الحركة

Looks

المظاهر

Sound

الأصوات

Pen

القلم

Events

الأحداث

Contrl

التحكم

Sensing

التحسس

Operators

العمليات





Motion

الحركة

الأمر

الشرح

حركة الكائن

المثال

move steps


تحرك عدد من الخطوات



when space key pressed
move 10 steps

turn degrees

دوران إلى اليمين من 0 - 360 درجة



when right arrow key pressed
turn 30 degrees

point in direction

أشـر باتجاه الزاوية
90 = يمين
-90 = يسار
0 = للأعلى
180 = للأسفل



point in direction 90



point in direction -90



point in direction 0



point in direction 180

point towards

أشـر باتجاه.
اختر من القائمة



when clicked
forever
point towards mouse-pointer

go to x: y:

إذهب إلى.
حدد موقع السينات والصادات



when clicked
go to x: -175 y: -50
say Now I am at stage left for 2 secs

go to mouse-pointer

إذهب إلى.
اختر الكائن من القائمة

when space key pressed
move 10 steps

when this sprite clicked
go to Sprite2





Motion

الحركة

الأمر

الشرح

حركة الكائن

المثال

glide 1 secs to x: 100 y: 100

تحرك في زمن
معين إلى موقع
س و ص



```
when green flag clicked
go to x: -100 y: -50
glide 2 secs to x: -10 y y: 0
glide 2 secs to x: 20 y: 90
```

change x by 10

قم بتغيير الحركة
الأفقية بمقدار
محدد



```
when space key pressed
change x by 10
```

set x to 0

قم بتحديد الموقع
السيني (الأفقي)
بين 240 و -240



```
when green flag clicked
set x to -200
say I am at the left for 2 secs
set x to 200
say I am at the right for 2 secs
set x to 0
say I am at the center for 2 secs
```

change y by 5


قم بتغيير الحركة
العمودية بمقدار
محدد

```
when I receive message1
change y by 5
```

```
when I receive message1
change y by -5
```

set y to 0

قم بتحديد الموقع
الصادي (العمودي)
بين 180 و -180



```
when green flag clicked
set y to 100
say I am near the top for 2 secs
set y to -100
say I am near the bottom for 2 secs
set y to 0
say I am near the middle for 2 secs
```

if on edge, bounce

إذا ارتطم بالإطار ،
ارتد



```
when green flag clicked
forever
move 10 steps
if on edge, bounce
```





Motion

الحركة

الأمر

الشرح

حركة الكائن

المثال

set rotation style left-right

إختيار وضع الإلتفاف.
إختر من القائمة



```

when clicked
set rotation style left-right
forever
  move 10 steps
  if on edge, bounce

```

x position

الموقع الحالي
للكائن أفقيا


```

when clicked
forever
  say x position

```

y position

الموقع الحالي
للكائن عموديا



```

when clicked
forever
  if y position < -150 then
    say Game Over!

```

direction

الإتجاه (زاوية)

```

when clicked
forever
  if direction < 0 then
    say Heading up!

```





Looks

المظاهر

الأمر

الشرح

حركة الكائن

المثال

say [] for [] secs

إظهار (كلام)
لمدة محددة



say []

إظهار (كلام) لمدة
غير محددة




think [] for [] secs

إظهار (تفكير) لمدة
محددة



think []

إظهار (تفكير) لمدة غير
محددة



show

إظهار الكائن


hide

إخفاء الكائن



switch costume to costume2

تغيير لمظهر معين
لنفس الكائن باستخدام
القائمة





Looks

المظاهر

الأمر

الشرح

حركة الكائن

المثال

next costume

تغيير للمظهر الثاني
لنفس الكائن بشكل
تلقائي



switch backdrop to backdrop1

تغيير الخلفية إلى
الخلفية المحددة من
القائمة



switch backdrop to night city



switch backdrop to night city with street

set color effect to 0

تحديد (إختر من القائمة)
التأثير بمقدار



clear graphic effects


قم بحذف التأثيرات
على الكائن



now



1 second later




change size by

تغيير حجم الكائن
بمقدار معين



set size to 100 %

تحديد حجم الكائن
برقم محدد



set size to 50 %

set size to 200 %

set size to 100 %





Looks

المظاهر

الأمر

الشرح

حركة الكائن

المثال

go to front

إظهار الكائن في المقدمة



تغيير (أختر من القائمة) التأثير بمقدار


change color effect by 25



```
repeat 10
  change color effect by 25
  wait 1 secs
```

go back 1 layers

الرجوع عدد من معين من الطبقات



go back 1 layers

costume #

رقم المظهر لكائن محدد




The number next to the costume in the list is the costume #

```
next costume
if costume # = 1 then
  play sound meow
if costume # = 2 then
  play sound dog1
```

backdrop name

إستخدام اسم الخلفية الحالي لأغراض مختلفة



```
switch backdrop to backdrop1
if backdrop name = moon then
  play sound pop
```





Looks

المظاهر

الأمر


الشرح

حركة الكائن

المثال

size

إستخدام الحجم
الحالي لأغراض
أخرى



```
forever
if size > 200 then
  set size to 100 %
  change size by 10
```





Sound

الأصوات

الأمر

الشرح

حركة الكائن

المثال

play sound meow

تشغيل الصوت.
أختر من القائمة

```
when space key pressed  
play sound meow
```

play sound dog1 until done

تشغيل الصوت لحد
انتهاء الشريط الصوتي.
أختر من القائمة

```
when space key pressed  
play sound dog1 until done  
say Do you know which animal made this sound ?
```

stop all sounds

أوقف جميع
الأصوات

```
when I receive message1  
stop all sounds  
play sound dog1
```

play drum 5 for 0.2 beats

شغل الطبل (معين)
لمدة عدد وحدات إيقاع

```
repeat 10  
play drum 7 for 0.2 beats  
play drum 5 for 0.2 beats
```

rest for 0.25 beats

إسترح لمدة (محددة)
وحدة إيقاع

```
play note 60 for 0.5 beats  
play note 62 for 0.5 beats  
rest for 0.20 beats  
play note 62 for 0.5 beats
```

play note 62 for 0.5 beats

عزف النوتة (معينة) لمدة
(محددة) وحدة إيقاع

```
repeat 3  
play note 60 for 0.5 beats  
play note 62 for 0.5 beats  
play note 62 for 0.5 beats
```

set instrument to 1

إجعل الآلة هي
(إختر من القائمة)

```
when space key pressed  
set instrument to 1  
play note 60 for 0.5 beats  
set instrument to 21  
play note 55 for 0.5 beats
```





Sound

الأصوات

الأمر

الشرح

حركة الكائن

المثال

change volume by -10

غير شدة الصوت
بمقدار محدد

```

when clicked
set volume to 100 %
forever
  play sound door creak until done
  change volume by -10
  if volume = 0 then
    stop other scripts in sprite

```

set volume to 100 %

ضبط شدة الصوت
بمقدار محدد

```

when clicked
set volume to 100 %
forever
  play sound door creak until done
  change volume by -10
  if volume = 0 then
    stop other scripts in sprite

```

volume

إستخدام شدة
الصوت الحالية

```

when clicked
set volume to 100 %
forever
  play sound door creak until done
  change volume by -10
  if volume = 0 then
    stop other scripts in sprite

```

change tempo by 20

غير سرعة الأداء
بمقدار محدد

```

when clicked
set tempo to 60 bpm
forever
  play note 60 for 0.5 beats
  change tempo by 20
  if tempo = 500 then
    stop this script

```





Sound

الأصوات

الأمر

الشرح

حركة الكائن

المثال

إجعل سرعة الأداء
مساويا (عدد معين)
من وحدات الإيقاع في
الدقيقة

set tempo to 60 bpm

```
when clicked
set tempo to 60 bpm
forever
  play note 60 for 0.5 beats
  change tempo by 20
  if tempo = 500 then
    stop this script
```

tempo

سرعة الأداء

```
when clicked
set tempo to 60 bpm
forever
  play note 60 for 0.5 beats
  change tempo by 20
  if tempo = 500 then
    stop this script
```





وَلَيْسَ أَخُو عِلْمٍ كَمَنْ هُوَ جَاهِلٌ
صَغِيرٌ إِذَا التَّمَّتْ عَلَيْهِ الْجَافِلُ
كَبِيرٌ إِذَا رُدَّتْ إِلَيْهِ الْمَحَافِلُ

تَعْلَمُ فَلَيْسَ الْمَرْءُ يُولَدُ عَالِمًا
وَإِنَّ كَبِيرَ الْقَوْمِ لَأَعْلَمُ عِنْدَهُ
وَإِنَّ صَغِيرَ الْقَوْمِ إِنْ كَانَ عَالِمًا

الشافعي





Pen

القلم

الأمر

الشرح

حركة الكائن

المثال

set pen color to

تحديد لون الخط
بعدد معين



```
pen down
repeat 100
  move 2 steps
  set pen color to y position
```

change pen shade by

تغيير مقدار تظليل
الخط بعدد معين



```
when clicked
  pen down
  set pen size to 10
  set pen color to blue
  set pen shade to 0
  repeat 100
    move 2 steps
    change pen shade by 1
```

set pen shade to


تحديد مقدار تظليل
الخط بعدد معين



```
pen down
set pen size to 10
set pen color to blue
set pen shade to 0
repeat 100
  move 2 steps
  change pen shade by pick random 1 to 100
```

change pen size by

تغيير سمك الخط
بمقدار معين



```
set pen size to 1
pen down
repeat 100
  move 5 steps
  turn 3 degrees
  change pen size by 1
```

set pen size to

تحديد سمك الخط
بعدد معين



```
pen down
set pen color to blue
set pen size to 20
move 50 steps
```





Pen

القلم

الأمر

الشرح

حركة الكائن

المثال

clear

مسح تأثير القلم

stamp

طباعة الكائن




```

clear
repeat 8
  move 70 steps
  turn 45 degrees
  stamp

```

pen down

أمر بداية الكتابة



```

pen down
move 10 steps

```

pen up

أمر نهاية الكتابة



```

pen down
move 80 steps
pen up
move 80 steps

```

set pen color to

إختيار لون الحبر



```

pen down
set pen color to
move 80 steps
set pen color to
move 80 steps

```

change pen color by 10

تغيير لون الحبر بمقدار محدد



```

when clicked
  pen down
  set pen color to
  repeat 200
    move 1 steps
    change pen color by 1

```





Events

الأحداث

الأمر

الشرح

حركة الكائن

المثال

when clicked

عند النقر على العلم الأخضر

```
when clicked
go to x: 0 y: 0
forever
  move 10 steps
  turn 15 degrees
```

when space key pressed

عند النقر على زر محدد من القائمة

```
when m key pressed
move 10 steps
```

when this sprite clicked

عند النقر على كائن معين

```
when this sprite clicked
play sound meow
```

when backdrop switches to Kitchen

عند تغيير الخلفية المحددة من القائمة



```
switch backdrop to backdrop1
```

```
when backdrop switches to Kitchen
say Now I'm in the Kitchen! for 4 secs
```

when video motion > 10

عندما تكون (شدة الصوت أو المؤقت أو حركة الفيديو) أكبر من عدد معين




```
when timer > 10
say Times up! for 2 secs
```



```
when video motion > 10
change color effect by 25
```





Events

الأحداث

الأمر

الشرح

حركة الكائن

المثال

broadcast

قم بإرسال رسالة
معينة (قم بكتابة
اسمها او اختر من
القائمة)

when I receive

عند استلام رسالة
محددة

when I receive message1

change y by 40

wait 0.5 secs

change y by -40

wait 0.5 secs

broadcast and wait

قم بإرسال الرسالة
المحددة أو المكتوبة
وانتظر لحد انتهاء الأمر

broadcast Jump and wait

say That was fun !

when I receive Jump

change y by 50

wait 1 secs

change y by -50





Control

التحكم

الأمر

الشرح

حركة الكائن

المثال

wait 1 secs

إنتظر عدد من
الثواني

```
move 10 steps
wait 1 secs
move 10 steps
```

repeat 10

تكرار الأوامر عدد
محدد من المرات

```
repeat 10
  move 10 steps
  wait 0.5 secs
```

forever

تكرار الأوامر
باستمرار

```
forever
  move 10 steps
  wait 0.5 secs
```

if then

الجملة الشرطية:
إذا حدث الامر
المحدد

```
forever
  if score > 10 then
    say you win
```

if then
else

الجملة الشرطية:
إذا حدث الامر
المحدد، وإلا فقم
بالتالي

```
forever
  if touching color ? then
    say We found water
  else
    say
```

wait until

الجملة الشرطية:
إنتظر حتى يحدث
الأمر المحدد

```
wait until touching Dog1 ?
  play sound dog1
  turn 15 degrees
  move 50 steps
```





Control

التحكم

الأمر

الشرح

حركة الكائن

المثال




الجملة التكرارية:
كرر حتى يحدث الأمر المحدد



أوقف (الكل)
او المقطع البرمجي الحالي
أو المقاطع الأخرى في
الكائن



عندما يبدأ استنساخ
كائن معين



أنشئ استنساخ للكائن
المحدد من القائمة



إمسح الكائن المستنسخ





Sensing

التحسس

الأمر


الشرح

حركة الكائن

المثال

touching ?

إذا لمس (أختر من القائمة)




touching color ?

إذا لمس لون محدد (إختر من القائمة)



color is touching ?

إذا لمس لون محدد لون آخر تم تحديده.



distance to

المسافة بين الكائن وشي آخر (أختر من القائمة)



ask and wait

إسأل سؤال وانتظر





Sensing

التحسس

الأمر

الشرح

حركة الكائن

المثال

answer

الإجابة

```

ask Do you want an apple or orange? and wait
if answer = apple then
  switch costume to holding an apple
if answer = orange then
  switch costume to holding an orange

```

key space pressed?

ضغط زر المسافة؟

```

forever
if key space pressed? then
  move 2 steps

```

mouse down?

ضغط زر الفأرة؟


```

if mouse down? then
  play drum 1 for 0.25 beats

```

mouse x

موقع مؤشر الفأرة أفقيا



```

forever
set whirl effect to mouse x

```

mouse y

موقع مؤشر الفأرة عموديا



```

forever
set ghost effect to mouse y

```

loudness

شدة الصوت

```

forever
set size to loudness %

```





Sensing

التحسس

الأمر

الشرح

حركة الكائن

المثال

video motion on this sprite

حركة الفيديو مع الكائن
المحدد

video motion on Stage

حركة الفيديو مع
المنصة

video direction on this sprite

إتجاه الفيديو مع الكائن
المحدد

video direction on Stage

إتجاه الفيديو مع
المنصة

```
when green flag clicked
forever loop
  if video motion on this sprite > 0 then
    play sound recording1
    hide
    wait 1 secs
    show
```


turn video on

أمر تشغيل الكاميرا

```
when space key pressed
turn video off
switch backdrop to backdrop1
```

set video transparency to 50 %

شفافية الفيديو
المعروض في المنصة




```
when I receive message1
set video transparency to 50 %
switch backdrop to Kitchen
```

timer

المؤقت

```
reset timer
wait until timer > 5
forever loop
  point towards Dog2
  move 10 steps
```





Sensing

التحسس

الأمر

الشرح

حركة الكائن

المثال

reset timer

تصفير المؤقت

```

reset timer
wait until timer > 5
forever
  point towards Sprite1
  move 5 steps

```

x position of Sprite1

(الموقع والحجم والإسم
وشدة الصوت ورقم
المظهر) لكائن محدد

```

when clicked
if costume name of Robot1 = mouth closed then
  play sound door creak

```

current minute

في هذه اللحظة (ثانية،
دقيقة، ساعة، يوم الأسبوع،
تاريخ اليوم، الشهر ، السنة)

```

say What is year is it ? for 2 secs
say current year for 2 secs

```

days since 2000

عدد الأيام من عام
2000

username

إسم المستخدم

```

forever
  if mouse down? then
    say username for 2 secs

```





Operators

العمليات

الأمر

الشرح

حركة الكائن

المثال



عملية جمع

```
say Watch me add 3 plus 4 equals... for 2 secs  
say 3 + 4 for 2 secs
```



عملية طرح

```
say Watch me subtract 3 minus 4 equals ... for 2 secs  
say 3 - 4 for 2 secs
```



عملية ضرب

```
say Watch me multiply 3 times 2 equals... for 2 secs  
say 3 * 2 for 2 secs
```



عملية قسمة

```
say Watch me 3 divide over 4 equals ... for 2 secs  
say 3 / 4 for 2 secs
```

pick random 1 to 10

إختر رقم عشوائي
من () إلى ()

```
set x to pick random -240 to 240  
set y to pick random -180 to 180
```



أصغر من

```
forever  
if distance to mouse-pointer < 10 then  
  move 3 steps
```



يساوي

```
forever  
if direction = 0 then  
  switch costume to costume2
```





Operators

العمليات

الأمر

الشرح

حركة الكائن

المثال



أكبر من

```


forever
  if y position > 200 then
    point in direction 180
  
```



(الأمر) و (الأمر)

```

forever
  if touching color [ ] and touching mouse-pointer ? then
    turn 15 degrees
  
```



(الأمر) أو (الأمر)

```

forever
  if touching edge ? or touching Sprite2 ? then
    play drum 2 for 0.25 beats
  
```



ليس (الأمر المختار)

```

forever
  if not touching Sprite2 ? then
    point towards mouse-pointer
    move 10 steps
  
```



أربط (كلمة) بـ (كلمة)



```

set things to moon
say join good night, things
  
```



```

say join I am a cat
  
```





Operators

العمليات

الأمر

الشرح

حركة الكائن

المثال


letter **1** of world

الحرف (رقم معين)
من (الكلمة)

```
set counter to 0
repeat length of mississippi
  say letter counter of mississippi for 1 secs
  change counter by 1
say letter 1 of world
```

length of world

عدد أحرف ()



say length of elephant

mod

باقي قسمة
(عدد) على (عدد)

```
forever
  move 1 steps
  if x position mod 10 = 0 then
    change y by 10
    wait 0.3 secs
    change y by -10
```

round

قرب (عدد)

```
forever
  say round timer
```

sqrt of 9

إختر عملية حسابية من
القائمة) لعدد معين.


```
sqrt of 9
abs
floor
ceiling
sqrt
sin
cos
tan
asin
acos
atan
ln
log
e ^
10 ^
```





رأى بالأمس جداً!!!!!!!!!!!!!! أيها المبرمج العبقرى... ممتناً!!!!!! از
بالفعل .. أنت تستحق أن تكون من ضمن مبرمجى المستقبل


واصل الطريق القرية الهندسية





نحن في القرية الهندسية نلم بجيل من الأطفال والشباب يجد متعته في العمل وإكتساب المعرفة. نلم بمستقبل مشرق ينافس فيه أبناءنا عباقرة العالم ويتفوقون عليهم بجد وشغف ولهذا نعمل جاهدين على تقديم العلوم بطريقة مسلية تنافس البدائل المتوفرة من ألعاب الفيديو ومشاهدة التلفاز. العالم من حولنا يتطور بسرعة رهيبه في مختلف مجالات التكنولوجيا الرقمية والإلكترونية ومن واجبنا أن نسعى لمواكبة التطور والحدثة ليس بالقييل والقال وإنما بالعلم والعمل. ومعا يمكننا أن نصنع التغيير !

فهد بن سعيد بن محمد السيابي
الرئيس التنفيذي للقرية الهندسية





المراجع

موقع سكراتش التعليمي
<https://scratch.mit.edu>

دورة البرمجة للأطفال
القرية الهندسية - صيف 2015

تم تصميم هذا الكتاب بالإعتماد على نتائج تجربة ميدانية واسعة مع مجموعة كبيرة من الأطفال ونخص بالشكر المدربين الرائعين:

إسحاق الراشدي
الوضاح البطاشي
سليم الراشدي
طارق المعمرى
ماجد العمرى
محمود الإسماعيلي
مهند الرحيلي

وشكر خاص للطلبة المبدعين الذين شاركوا في تجربة الكتاب ونخص بالذكر :

عيسى العامري
خليفة العامري
طارق الحبسي
حمود المعمرى
مجد الهنائي
أبها الحارثي
وائل الرحبي
محمد الحارثي
العنود السيابي

آية الشيدي
ميرة المسكري
ناصر المسكري
أبرار العامري
قبس العامري
إلياس السعيدى
ميماس الرحبي
إيلاف الرحبي
حمد الراشدي





مؤلفاتنا الأخرى



تروس في عالم الإلكترونيات
تعلم الإلكترونيات في 10 أيام




احترف الأردوينو
في 10 أيام



احترف الرازبيري باي
في 10 أيام





نرجو أن نكون قد استطعنا افادتكم وامتاعكم في عالم
برمجة سكراتش. ولمزيد من المعلومات وللتحقق من اجاباتكم زوروا موقعنا
على الإنترنت www.ev-center.com ، كما يسعدنا قراءة آرائكم
وتعليقاتكم في حساباتنا على شبكات التواصل الإجتماعي عبر
الوسم #مبرمجو_المستقبل



@EngineeringOman



+968 22027662



Engineering Village



+968 97377800



EngineeringVillageEV



www.ev-center.com



@EngineeringVillage



school@ev-center.com



تم بحمد الله

* كل شخص عليه أن يتعلم برمجة الحاسوب .. لأنه يعلمك كيف تفكر ،، ” ستيف جوبز“

* بيل جيتس | مؤسس ميكروسفت | تعلم البرمجة بعمر 13 عام

* مارك زوكربيرج | مؤسس الفيس بوك | بدأ باستخدام الحاسوب في الصف السادس

* جاك زورسي | مؤسس شبكة تويتر | أهداه والداه حاسوب وهو بعمر 9 سنوات

لم يكونوا يملكون العبقرية أو الذكاء الخارق وإنما بحثوا عن المرح والمتعة في أماكن الابتكار .

البرمجة هي لغة العصر لخدمة الإنسانية ومساعدة الناس ، هي طريقة التفكير والتنوير لعالم اليوم ويجب أن تبدأ بها منذ صغرك ! سواء أردت أن تغير العالم أو أن تملك الكثير من المال .

في العشر السنوات القادمة سيحتاج العالم إلى 1.4 مليون مبرمج لخدمة البشرية ، ولا يوجد في العالم سوى 400 ألف مبرمج حالياً . العالم يحتاجنا ، فهيا بنا نبرمج .



لنمرح بتعلم مهارات البرمجة مع سكراتش

#مبرمجوالمستقبل